



Notes on Maple for Calculus

Author: Fei Ye

Date: 2020-03-12

Version: 1.0

Contents



Introduction	1
1 Basics in Maple	2
Part 1 - Calculus I	9
2 Limits	9
3 Derivatives	16
4 Application of Differentiation	28
5 Integrals	39
6 Application of Integrals I	43
Part 2 - Calculus II	47
7 Applications of Integrals II	47
8 Calculus of Inverse Functions	49
9 Techniques of Integration	58
10 Further Applications of Integration	60
11 Infinite Sequences and Series	62

Introduction

This is a book written for Maple labs for Calculus I and II. The companion textbook is Stewart's Calculus book.

Maple 2018 was used for Calculus II and Maple 2019 was used for Calculus I.

The resource can be found at <https://github.com/fyemath/maple4calc>.

Comments and suggestions are very welcome.

This work is licensed under a [Creative Commons Attribution-NonCommercial-ShareAlike 4.0 International License](#).




by-nc-sa license icon

Chapter 1 Basics in Maple


1.1 What should I do after I opened Maple

Once you opened Maple, you will see the following Maple Start document.



Maple start page screenshot

- If you already know what you want to do, then you may open a new document by clicking **New Document** icon in the start document. The following shows what an new (empty document) looks like.




Maple new document page screenshot


In this new document you may type in text under **Text** mode or evaluate a Maple syntax in the **Math** mode. (See the following picture).

- If you want to explore some featured sample documents, you may go to **Start.mw** document and click on different icons to open a new document.
 - You may always reopen the start page by clicking the home icon to reopen the start page.
- For Calculus, the most useful document is **Calculus**.


If you click the **Calculus** icon on the Start page and click **OK**, you will see the following document.




Maple text and Math modes screen shot



Maple reopen start page screenshot



Maple start page screen shot with Calculus highlighted



Maple Calculus document page screen shot

1.2 Basic Operators

The first thing to know when learn something new is where and how to get help. In Maple, you may simply type

`?keyword`

to open a help page (a new window).

For example, the command `?operators` will lead you to descriptions of arithmetic operators in Maple.

	addition	subtraction	multiplication	division	exponentiation
Maple Operators	+	-	*	/	^
In writing	$x + 2$	$a - b$	$2x$	$\frac{p}{q}$	b^5
In Maple	<code>x+2</code>	<code>a-b</code>	<code>2*x</code>	<code>p/q</code>	<code>b^5</code>

1.3 How to define a function

A function is an assignment, for a given input x , we assignment an output y under a certain rule. Maple take this idea to define functions. The command to define a function has the following form.

`function name := independent variable -> function rule`

Here `:=` means “defined/assigned to be” and `->` may be understood as “plug in”.

Example 1.1 Define the following function in Maple and find the value $f(0.999)$.


$$f(x) = \frac{x}{x-1}$$

Solution The function name is f , the independent variable is x and the function rule is $\frac{x}{x-1}$. So the function can be defined in Maple by the following command.

`f:=x->x/(x-1)`

Once the function is define, you may find the function value by the command `f(0.999)`.


Remark The assignment operator `:=` to the left-hand side the value of the right-hand side. The left-hand side normally is a name and the right-hand side is a value or expression.

 **Exercise 1.1** Define the following function in Maple and find the value $f(2.0001)$.

$$g(x) = \frac{x^3}{(x-2)^2}$$

1.4 Initially known mathematical functions

Maple has many predefined functions which can be used to create new functions. To see all initially known mathematical functions in maple, you may use the help command `?functions` and click the hyperlinked “initial functions” in the description shown in the new window.



Maple function help page screenshot

Some frequently used functions are listed in tables below.

absolute value	square root	n-th root	natural exponential	logarithmic	
<code>abs()</code>	<code>sqr()</code>	<code>surd(,n)</code>	<code>exp()</code>	<code>log(), log[b](), ln()</code>	

sine	cosine	tangent	cotangent	secant	cosecant
<code>sin()</code>	<code>cos()</code>	<code>tan()</code>	<code>cot()</code>	<code>sec()</code>	<code>csc()</code>

inverse sine	inverse cosine	inverse tangent	inverse cotangent	inverse secant	inverse cosecant
<code>arcsin()</code>	<code>arccos()</code>	<code>arctan()</code>	<code>arccot()</code>	<code>arcsec()</code>	<code>arccsc()</code>

Another initially know function that we will use is the piecewise function.

`piecewise(condition1, expression1, condition2, expression2, expression3)`


Example 1.2 Define the following function in Maple and evaluate $pwf(3)$

$$pwf(x) = \begin{cases} \sqrt{\sin(x)} & x < -1 \\ \frac{\sqrt[3]{x}}{|x+2|} & -1 \leq x < \pi \\ \ln(e^x + 2) & \text{otherwise.} \end{cases}$$

Solution The function can be defined by the following command.

```
pwf:=x->piecewise(x<-1, sqrt(sin(x)), x>=-1 and x<Pi,
surd(x, 3)/abs(x+2), ln(exp(x)+2))
```

The value $\text{pwf}(3)$ can be obtained by the command $\text{pwf}(3)$.

 **Exercise 1.2** Define the following function in Maple and evaluate $q(1)$

$$q(x) = \begin{cases} |x-2|/\sqrt{x} & x > \frac{\pi}{2} \\ (x-1)\tan(x) & 0 \leq x < \frac{\pi}{2} \\ \sqrt[5]{\log_2(1+e^x)} & \text{otherwise} \end{cases}$$

1.5 Plot functions

In Maple, you may plot a single variable function easily using the command

```
plot(expression, domain, options)
```

or plot several single variable functions together using


```
plot([expression1, expression2], domain, options)
```

In the command, options may be omitted, but the domain must be given. To see details about available options, you may run the command `?plot` in Maple.


Example 1.3 Plot the functions $f(x) = x^2$ in red and $l(x) = 2x + 1$ in blue over the domain $[-1, 2]$.


Solution Here are the command and the output

```
plot([x^2, 2*x+1], x=-1..2, color=[red, blue])
```



Screen shot of the output generated by plotting of two function

 **Exercise 1.3** Plot the piecewise function in Exercise 1.2 over the domain $[-2, 4]$.

 **Exercise 1.4** Plot the functions $f(x) = \ln(x + 5)$ and $g(x) = 3\cos(2x + 1) + 4$ over the domain $[-\pi, \pi]$.

Part 1 - Calculus I

Chapter 2 Limits

2.1 Understand limits using tangent lines

Intuitively, the limit of a function f at $x = a$ is a fixed value L that values of the function $f(x)$ approach as the values of x ($\neq a$) approach a .

The slope of a tangent line to the graph of a functions is a limit of slopes of secant lines. This can be visualized easily using Maple with the support of the package `Student[Calculus1]`.

Like predefined functions in Maple, package consists of predefine commands for Maple. The package `Student` serves for studying Calculus and other subject interactively. The subpackage `Student[Calculus1]` focus mainly on Calculus as the name indicated.

As different package has different focus and serves for different purpose, Maple won't load a specific package until you run the command `with(package_name)`. For example, the command `with(Student[Calculus1])` will load the subpackage `Calculus1`.


Example 2.1 Observe how do secant lines of the function $f(x) = x^3 - 2$ approach to the tangent line at $x = 1$.

Solution Load the `Student[Calculus1]` package using `with()`.

```
with(Student[Calculus1])
```

Use `TangentSecantTutor` from the loaded package to observe changes of secant lines.

```
TangentSecantTutor(x^3-1, x=1)
```

 **Exercise 2.1** Explore the package `Student`, in particular the subpackage `Student[Calculus1]`. You can use the command `?Student` to get help.

Find the slope of the tangent line to the function $f(x) = 2x^3 + \frac{1}{x^2}$ at $x = 1$ using the `TangentSecantTutor` command.

2.2 Estimate limits numerically or graphically

To estimate a limit $\lim_{x \rightarrow a} f(x)$ numerically, one may pick some values close to a and evaluate the function. In Maple, the calculation can be done by using the repetition statement `for counter in array do statement end to;`

Example 2.2 Estimate the limit $\lim_{x \rightarrow 0} \frac{\sin x}{x}$ by approximations.

Solution First, we pick some values close to 0, for example -0.01, -0.001, -0.0001, 0.01, 0.001, 0.0001 and assign them to an expression.

```
sq:=[-0.01, -0.001, -0.0001, 0.0001, 0.001, 0.01]
```

Now we find the function values using two new commands instead of defining the function a priori.

```
for t in sq do evalf(subs(x=t, sin(x)/x)) end do;
```

Graphs provide visual intuition which helps understand and solve problems. Recall, the command `plot(expression, domain, option)` produces a graph of the function defined by the expression over your choice of domain.


Example 2.3 Determine whether the limit $\lim_{x \rightarrow 0} \frac{1}{1-\cos x}$ exists.

Solution Apply the `plot` function to the expression over the domain $(-0.5, 0.5)$.

```
plot(1/(1-cos(x)), x=-0.5..0.5)
```

The graph shows that the function $y = \frac{1}{1-\cos x}$ goes to ∞ when x approaches 0. So the limit is an infinite limit.

 **Exercise 2.2** Estimate the limit $\lim_{t \rightarrow 0} \frac{1-\cos x}{x}$ numerically.

 **Exercise 2.3** Determine whether the limit $\lim_{x \rightarrow 1} \frac{\sin x}{|x-1|}$ exists using the graph.

2.3 Evaluate limits

Maple provides the following command to evaluate a limit

```
limit(function, position, direction)
```

The direction may be omitted when evaluating a two-side limit.

Example 2.4 Determine whether the limit $\lim_{x \rightarrow 0} \frac{|x|}{x}$ exists.

Solution You may find the left and right limits using the following commands.


```
limit(abs(x)/x, x=0, left);
limit(abs(x)/x, x=0, right);
```


It turns out that $\lim_{x \rightarrow 0} \frac{|x|}{x}$ does not exist because the left limit and the right limit are different.

Example 2.5 Evaluate the limit $\lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h}$, where $f(x) = \frac{1}{x}$.

Solution The limit can be obtained using the following command.

```
limit((1/(x+h)-1/x)/h, h=0)
```

 **Exercise 2.4** Determine whether the limit $\lim_{x \rightarrow 1/2} \frac{2x-1}{|2x^3-x^2|}$ exists.

 **Exercise 2.5** Evaluate the limit $\lim_{t \rightarrow 0} \frac{\sqrt{x+t} - \sqrt{x}}{t}$.

2.4 Learn limit laws using **LimitTutor**

Suppose the limits of two functions f and g at the same point $x = a$ exist (equal finite numbers). Then the limit operation commutes with addition/subtraction, multiplication/division and power. In Maple, you may use the command `LimitTutor(function, position, direction)`, which is again supported by the subpackage `Student[Calculus1]`, to learn how to evaluate a limit using limit laws and theorems.

Example 2.6 Evaluate $\lim_{x \rightarrow 0} \frac{1 - \cos x}{x}$.


Solution Load the subpackage `Student[Calculus1]` if it was not loaded.

```
with(Student[Calculus1])
```

Use the following command to learn how to evaluate the limit.

```
LimitTutor(2-x^3, x=2,'right')
```

You will see an interactive windows pop out. You can choose the see the procedure step-by-step.

 **Exercise 2.6** Evaluate $\lim_{x \rightarrow 0} \frac{(x+2)(\cos x - 1)}{x^2 - x}$ using `LimitTutor`.

2.5 Squeeze Theorem

Comparison is a very useful tool in problem solving. Squeeze theorem is such an example

Theorem 2.1 (Squeeze Theorem) Suppose that

$$f(x) \leq g(x) \leq h(x)$$

and

$$\lim_{x \rightarrow c} f(x) = L = \lim_{x \rightarrow c} h(x).$$

Then

$$\lim_{x \rightarrow c} g(x) = L.$$


Let's use Maple to understand the statement.


Example 2.7 Graph the functions $f(x) = -x$, $g(x) = x \cos \frac{1}{x}$ and $h(x) = x$ in the same coordinate system. What's the limit of $g(x)$ as x approaches 0.

Solution We use the `plot()` command to graph the functions together.

```
plot([-x, x*cos(1/x), x], x=-1..1, discont, color=[red, blue, red])
```

From the graph, we see that the $\lim_{x \rightarrow 0} f(x) = 0$ as it is squeezed by two limits which are both 0.

 **Exercise 2.7** Graph the functions $f(x) = -x$, $g(x) = x \sin \frac{1}{x}$ and $h(x) = x$ in the same coordinate system. What's the limit of $g(x)$ as x approaches 0.



Squeeze Theorem demonstration

2.6 Continuity

A function f is continuous at $x = a$ if $f(a)$ is defined and $\lim_{x \rightarrow a} f(x) = f(a)$. Intuitively, a function is continuous if the graph has no hole or jump.

Example 2.8 Use graph to determine if the function

$$f(x) = \begin{cases} x & x \leq -1 \\ 1/(x-1) & -1 < x < 1 \\ 3-x & 1 \leq x \leq 2 \\ \sin(x-2) + 1 & x > 2 \\ -2 & x = 2 \end{cases}$$

is continuous over $(-\infty, \infty)$. Find the discontinuities and verify them using the definition and properties of continuity.

Solution First we define the function.


```
f := x -> piecewise(x <= -1, x, -1 < x and x < 1, 1/(x - 1),
1 <= x and x < 2, 3 - x, 2 < x, sin(x - 2) + 1, -2)
```

We first check visually whether the graph has holes or jumps.

```
plot(f(x), discont=[showremovable], x=-5..5, smartview=true)
```

In the above command, the option `discont = [showremovable]` is used to show removable discontinuities.

From the graph, we can tell that the function has discontinuities which can also be found using Maple.



An example shows three types of discontinuities

```
discont(f(x), x)
```

Maple gives three discontinuities $\{-1, 1, 2\}$.

Let's first find limits at all three values.

```
discontset:=[-1, 1, 2];
for a in discontset do limit(f(x), x=a) end do;
```

You will find that limits do not exist at -1 and 1 . The limit at 2 is 1 . However, $g(2) = -2$. So f has three discontinuities at $x = -1$, $x = 1$ and $x = 2$.

 **Exercise 2.8** Determine if the function

$$f(x) = \begin{cases} -x^2 + 2 & x \leq 1 \\ \frac{1}{x-2} & 1 < x < 2 \\ 2 \cos x - 1 & \text{otherwise} \end{cases}$$

is continuous over $(-\infty, \infty)$. Find all discontinuities if they exist and verify them using the definition.

2.7 Continuity and Limit of a Composite Function

Continuous functions behave under composition. The composition of continuous functions is still continuous over its domain. More generally,

Theorem 2.2 Let f be a function continuous at $x = L$. Suppose that $\lim_{x \rightarrow c} g(x) = L$. Then

$$\lim_{x \rightarrow c} f(g(x)) = f(L).$$

In general, the limit does not commute with composition.

Example 2.9 Let

$$f(x) = x^3 \quad \text{and} \quad g(x) = \begin{cases} 1 & x = 0 \\ 2x - 1 & \text{otherwise} \end{cases}$$

Verify that

$$\lim_{x \rightarrow 0} f(g(x)) = f(\lim_{x \rightarrow 0} g(x)).$$

Is it true that

$$\lim_{x \rightarrow 0} g(f(x)) = g(\lim_{x \rightarrow 0} f(x))?$$

Solution We first define f and g in Maple

```
f:=x->x^2;
g:=x->piecewise(x=0, 1, 2x-1);
```

Evaluate limits for $f \circ g$.


```
limit(f(g(x)), x=0);
f(limit(g(x), x=0));
```

The results verify that the limit commutes with composition if the outside function is continuous.

Evaluate limits for $g \circ f$

```
limit(f(g(x)), x=0);
f(limit(g(x), x=0));
```

The results show that the limit may not commute with composition if the outside function is not continuous.

 **Exercise 2.9** Find three functions f, g, h and a value c such that $f(g(x))$ is continuous,

$$\lim_{x \rightarrow c} f(h(x)) = f(\lim_{x \rightarrow c} h(x))$$

and

$$\lim_{x \rightarrow c} h(g(x)) \neq h(\lim_{x \rightarrow c} g(x)).$$

2.8 Intermediate Value Theorem

A very important result about continuity is the intermediate value theorem (IVT for short).

Theorem 2.3 (Intermediate Value Theorem) Let f be a function continuous over the interval $[a, b]$. Suppose that $f(a) \neq f(b)$ and N is a number between $f(a)$ and $f(b)$. Then there exists a number $c \in (a, b)$ such that $f(c) = N$.

In particular, if $f(a)f(b) < 0$ then there exists a number c such that $f(c) = 0$.

Example 2.10 Determine whether the equation

$$\sin^2 x + 2x - 1 = 0$$

has the solution in $(-1, 1)$. Estimate the solution if it exists.

Solution We first use the left hand side of the equation to define a function $eq1$.

```
eq1:=x->sin(x)^2 + 2*x - 1
```

Now let's verify that $eq1$ is continuous over $[-1, 1]$ using the command `iscont()`.

```
iscont(eq1(x), x=-1..1)
```

The result is *True*. We may apply the IVT. Let's check if the value $eq1(-1) \cdot eq1(1) < 0$.

```
evalf(eq1(-1)*eq1(1))
```

Here `evalf()` convert the symbolic answer to the (approximate) numerical value.

Since the product is negative, applying the IVT, we know there exists a solution between $(-1, 1)$.

This can also be seen from the graph using the command.

```
plot(eq1(x), x=-1..1)
```

To estimate the solution, you may use the Maple command

```
evalf(solve(eq1(x) = 0, x))
```

or you may repeatedly apply IVT to find an approximate solution.

```
m:=10;
a:=-1;
b:=1;
for n from 1 to m do
    c[n] := (a + b)/2;
    if evalf(eq1(c[n])*eq1(a)) < 0 then
        b := c[n];
    else
        a := c[n];
    end if;
    evalf(eq1(c[n]));
end do;
```

 **Exercise 2.10** Find an integer k such that the equation

$$\cos^2 x + 3x - 2 = 0$$

has a solution in $(k, k + 1)$. Estimate the solution.

Chapter 3 Derivatives

3.1 Derivative Functions

What makes derivative so important in modern mathematics is the ideal of linearly approximating curves using tangent lines. Geometrically, the derivative of a function f at a point $x = a$ is the slope of the line tangent to f at $x = a$. Using limits (if it exists), the derivative is defined as

$$f'(a) := \lim_{x \rightarrow a} \frac{f(x) - f(a)}{x - a}.$$

Consider a as a variable, we may define a function called the derivative function. In terms of limits, the derivative function of a function f , denoted by f' is given by

$$f'(x) = \lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h}.$$

Using limit laws, we can show that differentiable functions are continuous.

Geometrically, the graph of differentiable function locally is flat without any hole or jump.

All **elementary functions** are differentiable over their domain. But there are also many functions which are not differential everywhere in their domain.

Example 3.1 Use the graph of the function $f(x) = |x - 1|$ to identify a x value where f is not differential. Verify your finding using the definition of differentiability.

Solution First plot the function.

```
f:= x-> abs(x);  
plot(f(x), x=-2..2);
```

You will see the function is not flat near $x = 1$. To verify that, we calculate the limit of the difference quotient for all x and then evaluate the resulting function at $x = 1$.

```
diffquot:=(f(x+h)-f(x))/h;  
derlimit:= limit(diffquot, h=0);  
evalf(subs(x=1, derlimit));
```

 **Exercise 3.1** Determine whether the function

$$f(x) = \begin{cases} x \sin\left(\frac{1}{x}\right) & x \neq 0 \\ 0 & x = 0 \end{cases}$$

is differential at $x = 0$ using the definition of the differentiability.

 **Exercise 3.2** Determine whether the function

$$f(x) = \begin{cases} x^2 \cos\left(\frac{1}{x}\right) & x \neq 0 \\ 0 & x = 0 \end{cases}$$

is differential at $x = 0$ using the definition of the differentiability.

 **Exercise 3.3** Using the definition to find the derivative for $y = \sin x$.

3.2 Calculating Derivatives

Calculating a derivative in Maple is as easy as calculating a limit. The command for differentiation is `diff(function, variable)`. For higher derivatives, you may simply repeat the variable or use `[variable$n]` to indicate the n times differentiation.

Example 3.2 Calculate the first derivative and the second derivative for the function

$$f(x) = \frac{2}{\sqrt{2\sin^2 x + 1}}.$$

Solution First represent the function rule simply by f .

```
f:=2/(sqrt(2*(sin(x))^2+1))
```

Calculate the first derivative and denote the derivative by $f1$.

```
f1:=diff(f, x);
```

Calculate the second derivative and denote the derivative by $f2$. You may use one of the following three commands.

```
f2:=diff(f1, x);
f2:=diff(f, x, x);
f2:=diff(f, [x$2]);
```

Maple also has a tutoring command for differentiation: `DiffTutor(function, variable)` which is again supported by the subpackage `Student[Calculus1]`. However, `DiffTutor` only works for the first derivative.

Example 3.3 Calculate the derivative for $g(x) = \frac{x^2-2x-x^{-3}}{\sqrt{x}}$ by hand and compare your calculation the the result given by `DiffTutor`.

Solution By hand, we may simplify the expression using rational exponents first and then apply derivative rules.

$$g'(x) = (x^{\frac{3}{2}} - 2x^{\frac{1}{2}} - x^{-\frac{7}{2}})' = \frac{3}{2}\sqrt{x} - \frac{1}{\sqrt{x}} + \frac{7}{2x^4\sqrt{x}} = \frac{3x^5 - 2x^4 + 7}{2x^4\sqrt{x}}.$$


To see the result from *DiffTutor*, we use the following command. Remember to load the `Student[Calculus1]` first.


```
with(Student[Calculus1]):
DiffTutor((x^2-2*x-x^(-3))/sqrt(x), x);
```


Here is how does the output look like


$$\begin{aligned}
& \frac{d}{dx} \left(\frac{x^2 - 2x - \frac{1}{x^3}}{\sqrt{x}} \right) \\
&= \frac{\frac{d}{dx} (x^2 - 2x^4 - 1) x^{7/2} - (x^2 - 2x^4 - 1) \frac{d}{dx} (x^{7/2})}{x^7} \quad [\text{quotient}] \\
&= \frac{\left(\frac{d}{dx} (x^2) + \frac{d}{dx} (-2x^4) + \frac{d}{dx} (-1) \right) x^{7/2} - (x^2 - 2x^4 - 1) \frac{d}{dx} (x^{7/2})}{x^7} \quad [\text{sum}] \\
&= \frac{\left(\frac{d}{dx} (x^2) + \frac{d}{dx} (-2x^4) \right) x^{7/2} - (x^2 - 2x^4 - 1) \frac{d}{dx} (x^{7/2})}{x^7} \quad [\text{constant}] \\
&= \frac{\left(\frac{d}{dx} (x^2) - 2 \frac{d}{dx} (x^4) \right) x^{7/2} - (x^2 - 2x^4 - 1) \frac{d}{dx} (x^{7/2})}{x^7} \quad [\text{constantmultiple}] \\
&= \frac{\left(-8x^3 + \frac{d}{dx} (x^2) \right) x^{7/2} - (x^2 - 2x^4 - 1) \frac{d}{dx} (x^{7/2})}{x^7} \quad [\text{power}] \\
&= \frac{(5x^4 - 8x^3) x^{7/2} - (x^2 - 2x^4 - 1) \frac{d}{dx} (x^{7/2})}{x^7} \quad [\text{power}] \\
&= \frac{(5x^4 - 8x^3) x^{7/2} - \frac{7(x^2 - 2x^4 - 1) x^{5/2}}{2}}{x^7} \quad [\text{power}] \\
&= \frac{d}{dx} \left(\frac{x^2 - 2x - \frac{1}{x^3}}{\sqrt{x}} \right) = \frac{(5x^4 - 8x^3) x^{7/2} - \frac{7(x^2 - 2x^4 - 1) x^{5/2}}{2}}{x^7} \\
&= \frac{d}{dx} \left(\frac{x^5 - 2x^4 - 1}{x^{7/2}} \right) = \frac{3x^5 - 2x^4 + 7}{2x^{9/2}}
\end{aligned}$$

The derivative of a function obtained by the DiffTutor command

 **Exercise 3.4** Calculate the first derivative for the function $y = 2x^{-1} - 3\sqrt{x}$ by hand and by Maple. Compare two results. Are they different? If so, can you explain the difference?

 **Exercise 3.5** Calculate the first derivative for the function $y = x \sin^{-1}(x)$ by hand and by Maple. Compare two results. Are they different? If so, can you explain the difference?

 **Exercise 3.6** Calculate the first derivative for the function $y = \frac{\cos(x) + \sin x}{x}$ by hand and by Maple. Compare two results. Are they different? If so, can you explain the difference?

 **Exercise 3.7** Calculate the 3-th derivative for the function $y = \sin x \cos x$ by hand and by Maple. Can you find a formula for the n -th derivative of the function?

3.3 Chain Rule

Let $f(x)$ and $g(x)$ be two differentiable functions. Then the derivative of the composite function $f(g(x))$ can be calculated using the following formula

$$(f(g(x)))' = f'(g(x))g'(x).$$

Why there is an extra factor $g'(x)$? This is mainly because $g(x+h) \approx g(x) + g'(x)h$.

Now let's use Maple to understand the chain rule. In Maple, the symbol for composition is \circ , that is, in Maple the composition $f \circ g$ is given by $\mathbf{f \circ g}$.

Since we will evaluate derivative functions, in addition to the command `diff(expression, variable)`, we will also use `D(function)(variable)` to find the derivative function. One major difference between those two commands is that `D` is designed to differentiate functions, whereas `diff` is for differentiating expressions.

Example 3.4 Let $f(x) = \sin x$, $g(x) = 2x$ and $F(x) = 2 \sin x \cos x$. Find $f'(g(x))$, $(fg)'(x)$, $(f \circ g)'(x)$ and $F'(x)$? Compare the derivatives and draw a conclusion.

Solution We first define the functions.


```
f:=x->sin(x);
```

Operator D vs. Function diff

Description
Examples

Description

- The **D** operator computes derivatives of operators, while **diff** computes derivatives of expressions.
- The argument and result of **D** are functional operators, while the argument and result of **diff** are expressions.
- The following diagram illustrates the transitions that occur using **D**, **diff**, **unapply**, and **function application**.



Examples

```
> D(sin)(x);
```

```
cos(x)
```

```
(1)
```

```
> diff(sin(x), x);
```

```
cos(x)
```

```
(2)
```

A screen shot from Maple shows a comparison between the commands **D** and **diff**

```
g:=x->2*x;
```

```
F:=x->2*sin(x)*cos(x);
```

Find the derivative functions

```
Der_fg:=D(f)(g(x)); # f'(g(x))
```

```
Der_fg:=diff(f(x)g(x), x); # (fg)'(x)
```

```
Der_fog:=D(f@g)(x); # (fog)'(x)
```

```
Der_F:=D(F)(x); # F'(x)
```

To compare the derivatives, we use the command *expand* and *simplify* to rewrite the expressions.

```
simplify(expand(Der_fg));
```

```
simplify(expand(Der_fg));
```

```
simplify(expand(Der_fog));
```

```
simplify(expand(Der_F));
```

From the outputs, we see that $(f \circ g)'(x) = F'(x)$. Why they are the same? This is because $\sin(2x) = 2 \sin x \cos x$!

```
f:=x->sin(x);
g:=x->2*x;
F:=x->2*sin(x)*cos(x);
expand(f(2*x)-F(x));

f:=x->sin(x)
g:=x->2*x
F:=x->2*sin(x)*cos(x)
0

Der_fg:=D(f)(g(x)); # f'(g(x))
Der_fg:=diff(f(x)g(x), x); # (fg)'(x)
Der_fog:=D(f@g)(x); # (fog)'(x)
Der_F:=D(F)(x); # F'(x)

Der_fg:=cos(2*x)
Der_fg:=2*cos(x)*x+2*sin(x)
Der_fog:=2*cos(2*x)
Der_F:=2*cos(x)^2-2*sin(x)^2

simplify(expand(Der_fg));
simplify(expand(Der_fg));
simplify(expand(Der_fog));
simplify(expand(Der_F));

2*cos(x)^2-1
2*cos(x)*x+2*sin(x)
4*cos(x)^2-2
4*cos(x)^2-2
```

```
□
```

An example about the chain rule solved using Maple

Remark You may also use `D(unapply(f(x)*g(x), x))(x)` to calculate the derivative of the product function $(fg)(x) = f(x)g(x)$.

The chain rule is almost unavoidable in calculation of derivatives. Sometimes, using chain rule will simplify the calculation.

Example 3.5 Consider the function $f(x) = \frac{1}{\sin x + \cos x}$.

1. Find the derivative function $f'(x)$ using **DiffTutor**. Which rule of derivative was applied first?
2. Find the point where the tangent line of f is horizontal over the domain $(-\frac{\pi}{2}, \frac{\pi}{2})$.
3. Find an equation of the tangent line at $(0, 1)$.
4. Plot the tangent line and the function together over the domain $(-\frac{1}{2}, \frac{1}{2})$.

Solution You may use quotient rule to find the derivative. However, the chain rule may be a better choice because $\frac{1}{\sin x + \cos x} = (\sin x + \cos x)^{-1}$. Let's try *DiffTutor*.

```
restart; # Use `restart` to clear the internal memory.
with(Student[Calculus1]);
f:=1/(sin(x)+cos(x));
DiffTutor(f, x);
```

You see that the chain rule was applied first.

$$\begin{aligned}
 & \frac{d}{dx} \left(\frac{1}{\cos(x) + \sin(x)} \right) \\
 &= \left(\frac{d}{d_{X0}} \left(\frac{1}{_{X0}} \right) \right)_{X0 = \cos(x) + \sin(x)} \frac{d}{dx} (\cos(x) + \sin(x)) \quad [chain] \\
 &= \left(\frac{d}{d_{X0}} \left(\frac{1}{_{X0}} \right) \right)_{X0 = \cos(x) + \sin(x)} \left(\frac{d}{dx} \cos(x) + \frac{d}{dx} \sin(x) \right) \quad [sum] \\
 &= \left(\frac{d}{d_{X0}} \left(\frac{1}{_{X0}} \right) \right)_{X0 = \cos(x) + \sin(x)} \left(-\sin(x) + \frac{d}{dx} \sin(x) \right) \quad [cos] \\
 &= \left(\frac{d}{d_{X0}} \left(\frac{1}{_{X0}} \right) \right)_{X0 = \cos(x) + \sin(x)} (\cos(x) - \sin(x)) \quad [sin] \\
 &= -\frac{\cos(x) - \sin(x)}{(\cos(x) + \sin(x))^2} \quad [power] \\
 & \frac{d}{dx} \left(\frac{1}{\cos(x) + \sin(x)} \right) = -\frac{\cos(x) - \sin(x)}{(\cos(x) + \sin(x))^2}
 \end{aligned}$$

A example that chain rule was applied first by DiffTutor

To find the points where the tangent line is horizontal, we need to solve the equation $f'(x) = 0$.

```
solve(diff(f, x)=0)
```

To find the tangent line, we find the derivative $f'(0)$. One way is to use *unapply*.

```
unapply(diff(f, x), x)(0)
```

You may also use *eval(expression, variable=value)*.

```
eval(diff(f, x), x=0)
```


Since $f'(0) = -1$, the tangent line at $(0, 1)$ is defined by $y = -x + 1$.





```
y:= -x+1
```

Now let's verify visually that $y = -x + 1$ is the tangent line.

```
plot([f, y], x=-1/2..1/2)
```

 **Exercise 3.8** Let $f(x) = \sin x$, $g(x) = \frac{\pi}{2} - x$ and $h(x) = \cos x$. Find $f'(g(x))$, $(fg)'(x)$, $(f \circ g)'(x)$ and $h'(x)$. Compare the derivatives and draw a conclusion.

 **Exercise 3.9** Let $f(x) = x^2$ and $g(x) = \cos x$ and $h(x) = \frac{1}{2}(\cos(2x) + 1)$. Find $f'(g(x))$, $(fg)'(x)$, $(f \circ g)'(x)$ and $h'(x)$. Compare the derivatives and draw a conclusion.

 **Exercise 3.10** Consider the function $F(x) = \sin^2\left(\frac{\pi(x^2+1)}{3}\right)$.

1. Define three functions f , g and h so that $F(x) = f(g(h(x)))$.
2. Describe how the derivative function $F'(x)$ was calculated by **DiffTutor**.
3. Find the point where the tangent line of F is horizontal over the domain $(0, 1)$.
4. Find an equation of the tangent line of F at $(1, \frac{3}{4})$.
5. Plot the tangent line and the function together over the domain $(\frac{1}{2}, \frac{3}{2})$.

3.4 Implicit Differentiation

Implicit differentiation is an application of the chain rule. It provides a way to find the slope of a tangent line of a function implicitly defined by an equation, that is, the dependent variable is not isolated.

In Maple, there are two useful commands that help us understand implicit defined functions. The command `implicitplot(equation, domain, range, options)` supported by the package `plots` can be used to graph an implicitly defined function. The command `implicitdiff(function, dependent variable, independent variable)` can be used to find derivatives implicitly.

Example 3.6 Graph the function y of x defined by $x^2 + 2y^2 = 2x + 4y$ with its tangent line at $(2, 0)$ over the domain $(1, 3)$ and range $(-1, 2)$.

Solution Let's first assign an name to the equation. (Run *restart* first if x and y were previously used as names.)

```
restart;
eqnf:=x^2+2*y^2=2*x+4*y;
```

Now let's find the derivative function, and its value at $(0, 1)$.

```
D_eqnf:=implicitdiff(eqnf, y, x);
slope:=eval(D_eqnf, {x=2, y=0});
```

Let's define the tangent line

```
tangentline:= y=slope*(x-2)+0
```


Now we are ready to plot the function with the tangent line together.


```
with(plots);
implicitplot([eqnf, tangentline], x=1..3, y=-1..2, color=[red, blue]);
```


Remark Another way, which is more flexible, to put two or more graphs together is to use the command `display(graph1, graph2)` which is supported by the package `plots`. For example, the following commands will show two curves in a single picture.

```
with(plots):
g1 := implicitplot(x^2+y^2=1, x=-1..1, y=0..1):
g2 := plot(1-abs(x), x=-1..1):
display(g1, g2, title="Two Together");
```

Note that we use colone `:` at the end of a command to hide the output.

 **Exercise 3.11** For the ellipse $x^2 - xy + y^2 = 4$, find the locations of all horizontal tangent lines and plot them implicitly on the same graph as the relation over the interval $-3 \leq x \leq 3$ and $-3 \leq y \leq 3$.

 **Exercise 3.12** Find an equation of the line tangent to the curve $x^2 + (x - y)^3 = 9$ at $x = 1$. (You may want to use `solve` and `subs` to find the y when $x = 1$).

 **Exercise 3.13** Find all points (x, y) on the curve of $|x|^{2/3} + |y|^{2/3} = 8$ where lines tangent to the curve at (x, y) is perpendicular to the line $x - y = 1$. (Use `?solve` to learn how to solve a system of equations in Maple).

3.5 Rates of Change and Derivatives

Given a function $y = f(x)$, the average rate of change of f is the difference quotient

$$\frac{\Delta y}{\Delta x} = \frac{f(x + \Delta x) - f(x)}{\Delta x}.$$

The instantaneous rate of change is the limit $\lim_{\Delta x \rightarrow 0} \frac{\Delta y}{\Delta x}$ which is exactly the derivate of f . However, in science, economy and many other field, the same concept bears different names. For example, the velocity is the instantaneous rate of change of the position function; the marginal cost is the derivative of the cost function.

Now let's use Maple to help us understand such kind of application of derivatives.

Example 3.7 The position function of a particle moving along a straight line after t seconds is $s(t) = 2t^3 - 6t^2 - 18t + 1$ meters.

1. Find the time t that the particle is at the rest.
2. Find the distance the particle moved in the first 5 seconds.
3. Plot the $s(t)$, $v(t)$ and $a(t)$ for $0 \leq t \leq 5$ together.
4. When the particle is speeding up?
5. Confirm you answer in 3. using graph of the speed function.

Solution Let's first define if position function

```
s:= t->2*t^3-6*t^2-18*t+1
```

The particle at rest when the velocity is 0. To find t , we find $v(t)$ first and then solve t from $v(t) = 0$.

```
v:=D(s);
solve({v(t)=0, t>0}, t); # assuming t>0.
```

From the output, you will find that after 3 second the particle is temporarily at rest.

The particle moves forward if $v(t) > 0$ and backward if $v(t) < 0$. You may use the previous output or apply the command *solve* for those two inequalities. Either way, you will find from $t = 0$ to $t = 3$, $v(t) < 0$ and $v(t) > 0$ for $3 < t \leq 5$. So the total distance should be calculated as

```
totdist:= abs(s(3)-s(0))+abs(s(5)-s(3))
```

The acceleration function $a(t) = v'(t)$. To plot those three functions together, you may use *plot*.


```
a:=D(v);
plot([s(t), v(t), a(t)], t=0..5, color=[black, blue, red]);
```

The speed function of the particle is $|v(t)|$. When $v(t) > 0$ and $a(t) > 0$, the particle is speeding up. When $v(t) < 0$ and $a(t) < 0$, the particle is also speeding up. So the particle is speeding up if $v(t)a(t) > 0$. Similarly, the particle is slowing down if $v(t)a(t) < 0$.


```
solve({v(t)*a(t)>0, t>0}, t)
```

It shows that the particle is speeding up when $t < 1$ or $t > 3$. This can also be seen from the graph of the speed function.

```
plot(abs(v(t)), t=0..5)
```

 **Exercise 3.14** The position function of a particle moving along a straight line after t seconds is $s(t) = t^3 - 6t^2 - 15t + 2$ meters.

1. Find the time t that the particle is at the rest.
2. Find the distance the particle moved in the first 5 seconds.
3. Plot the $s(t)$, $v(t)$ and $a(t)$ for $0 \leq t \leq 5$ together.
4. When the particle is speeding up?
5. Confirm you answer in 3. using graph of the speed function.

 **Exercise 3.15** Suppose that the profit obtained from the sale of x calculators is given by $P(x) = -0.05x^2 + 4x + 25$. Use the marginal profit function to estimate the profit from the sale of the 51st calculator.

3.6 Related Rates

Stewart suggests the following general steps for solving problems about related rates.

1. Read the problem carefully. (What quantities are known? What is the unknown rate of change?)
2. Draw a diagram if possible.
3. Introduce notations. Assign symbols to all quantities that are functions of time.
4. Express the given information and the required rate in terms of derivatives.
5. Write an equation that relates the various quantities of the problem. If necessary, use the geometry of the situation to eliminate one of the variables by substitution.
6. Use the Chain Rule to differentiate both sides of the equation with respect to the shared independent variable, say t .
7. Substitute the given information into the resulting equation and solve for the unknown rate.

In Maple, the quantities that shared by the same independent variable should be written in function notations. See the following example for details.

Example 3.8 A spherical shaped balloon is inflated at a constant rate $2 \text{ cm}^3/\text{s}$. Find the relative rate growth of the diameter when its 10 cm wide.

Solution The volume of a sphere is given by $V = \frac{4\pi}{3}r^3$. In this equation, both V and r are functions of the time t . In Maple, we will use $V(t)$ and $r(t)$ for the volume and the radius.

```
eqn_v:=V(t)=4*Pi/3*(r(t))^3
```

We know that at a certain time $r(t) = 10/2 = 5$, and $\frac{d}{dt}V(t)|_{r(t)=5} = 2$. What is asked is $\frac{d}{dt}r(t)|_{r(t)=5}$. The two rates are related by the equation obtained by differentiating both side of the equation eqn_v .


```
D_eqn_v:=diff(eqn_v, t)
```


In the above equation, we need to solve the derivative of r with respect to t . Let's first simplify notations.

```
D_v:=diff(V(t), t);
D_r:=diff(r(t), t);
D_r:=solve(D_eqn_v, D_r);
```


Now we plugin $D_v=2$ and $r(t)=5$ to find the value for D_r .

```
subs({D_v=2, r(t)=5}, D_r) # or using eval(D_r, {D_v=2, r(t)=5})
```

 **Exercise 3.16** A cylindrical cup with radius 5 cm is being filled with coffee at a rate of $2 \text{ cm}^3/\text{s}$. How fast is the height of the coffee increasing?

 **Exercise 3.17** Two cars start moving from the same point. One travels north at 30 mph and the other travels west at 50 mph. At what rate is the distance between the cars changing one hour

later?

-  **Exercise 3.18** Sands dumping from a truck to the ground at a constant rate $5 \text{ m}^3/\text{s}$ is creating a circular cone. Find a relation between the rate of growth of the radius of the base and the rate of growth of the height of cone, when the the base is 6 m and the heigh is 4 m.

3.7 Linearizations

Let f be the function differentiable at $x = a$. The linearization of f at $x = a$ is defined to be the function $L(x) = f'(a)(x-a) + f(a)$. For any value b near a , the function value $f(b)$ is approximately the same as $L(b)$. This method is called linear approximation. The tangent line approximation is fundamental to almost every application of the derivative.

Example 3.9 Find the linearization $L(x)$ of the function $f(x) = \sqrt[3]{x+7}$ at $x = 1$ and use this linearization to estimate $f(0.99)$. How large is the error?

Solution We first find the slope fr the linearization which is the derivative $f'(1)$.

```
f:=x->surd(x+7, 3); # or f:=x->(x+7)^(1/3)
m:=simplify(D(f)(1)); # this is the slope (simplified).
```




Now we define the linearization $L(x)$ and estimate $f(0.99)$ using $L(0.99)$.

```
L:=x->m*(x-1)+f(1);
L(0.99);
```

The error may be calculated based on the output of the follow command.

```
f_L:=f(0.99)-L(0.99);
```

The results shows that the linear approximation is slightly over estimated with an error less than 4×10^{-7} .

-  **Exercise 3.19** Find the linear approximation $L(x)$ of the function $f(x) = \frac{\sqrt{x}}{x+1}$ at $x = 1$. Use this linearization to approximate $f(1.02)$.
-  **Exercise 3.20** Find the linear approximation $L(x)$ of the function $f(x) = \cos(2x)$ at $x = 0$. Use this linearization to approximate $f(0.1)$.
-  **Exercise 3.21** Find the linear approximation $L(x)$ of the function $f(x) = \sqrt{x^2 + 5}$ at $x = 2$. Use this linearization to approximate $f(2.03)$.

3.8 Newton's Method

In science and engineering, many problems may be eventually reduced to nonlinear equations which likely has no algebraic (analytic) solution. In such a situation, numerical solutions are hoped for applications. We've seen a method to find a numerical solution using the intermediate value theorem. However, it is not very effective. Using linearization, a root-finding algorithm

was developed by Newton and other mathematicians. The idea is to use the x -coordinates of tangent lines to approximate a root of an equation. Let's see an example first using the Maple command `NewtonMethod(function, starting point, options)` which is again supported by the subpackage `Student[Calculus1]`.

Example 3.10 Starting at $x = -1.5$, obtain a solution of $\sin x - \frac{x}{2} = 0$ by Newton's method.

Solution First load `Student[Calculus1]` use `with()`.

```
with(Student[Calculus1])
```

Now apply the `NewtonMethod` command

```
NewtonMethod(sin(x)-x/2, x=-1.5)
```

To see the graph, you may add the option `output=plot`.

```
NewtonMethod(sin(x)-x/2, x=-1.5, showroot=true, output=plot)
```

How does Newton's methods work? It's an iteration process. Consider the equation $f(x) = 0$. Suppose f is differentiable. To find a solution, we pick an initial value x_0 first. The x -value x_1 of the x -intercept of the linearization $L_0(x) = f'(x_0)(x - x_0) + f(x_0)$ should produce a value that is closer. The value x_1 is given by the formula

$$x_1 = x_0 - \frac{f(x_0)}{f'(x_0)}.$$

Iteratedly applying the above idea, we get the iteration formula

$$x_n = x_{n-1} - \frac{f(x_{n-1})}{f'(x_{n-1})}.$$

Remark How many times should be iterated? It depends on the sizes of acceptable errors and a maximum number of iteration.

If the x -values produced from two successive iterates is sufficiently small, say $|x_n - x_{n-1}| < e_1$, and the function value at $x = x_n$ is sufficiently small, say $f(x_n) < e_2$, where e_1 and e_2 are acceptable errors, then we may say $f(x) = 0$ has a solution at $x = x_n$.

If after iterated a maximum number of times and a solution was not found, then we may have to change the initial value or using other methods.

If it happens that $f'(x_n) = 0$, then the iteration process fails.

The above algorithm may be realized using the following Maple codes.

```
tol := 10^(-3);
N := 100;
f := x -> sin(x) - 1/2*x;
m := D(f);
newton := x -> evalf(x - f(x)/m(x)); # Newton's formula
x := -1.5;
```

```

for i to N do
  x := newton(x);
  if abs(x - newton(x)) < tol and abs(f(x)) < tol then
    break;
  eli i = N then
    error "Newton's method did not converge";
  end if;
end do;

print(x);




```

If you don't want to check the size of error, the code can be simplified using the composite symbol @ in Maple.

```

restart;
f := x -> sin(x) - 1/2*x;
m := D(f);
newton := x -> evalf(x - f(x)/m(x));
N:=100
x[N]:=evalf((newton@@N)(-1.5)); # @@N means that newton compose with itself N times.

```

-  **Exercise 3.22** Using Newton's Method to find solutions of the polynomial equation $x^3 - x^2 + 1 = 0$.
-  **Exercise 3.23** Using Newton's Method to find solutions of the equation $\cos x = \frac{x}{2}$.
-  **Exercise 3.24** Using Newton's Method to find the solution of the equation $\tan x - 2x = 0$ in the interval $[1, 2]$. Among 1 and 2, which is a better initial value? Why?

Chapter 4 Application of Differentiation

4.1 Maximum and Minimum Values

If f is continuous on a closed interval $[a, b]$, then f attains an absolute maximum value $f(c)$ and an absolute minimum value $f(d)$ at some numbers c and d in $[a, b]$.

A critical value of a function f is a number c in its domain such that either $f'(c) = 0$ or $f'(c)$ does not exist.

To find the absolute maximum and minimum values of f over $[a, b]$, we do the following.

1. Find the values of f at the critical values in $[a, b]$.
2. Find the values of $f(a)$ and $f(b)$.
3. The largest of the values from Steps 1 and 2 is the absolute maximum value; the smallest of these values is the absolute minimum value.

To find critical values, we solve the equation $f'(x) = 0$ and look at where $f'(x)$ is undefined. In Maple, you can use the command `CriticalPoints(function, variable)` which is supported by the subpackage `Student[Calculus1]`.

Example 4.1 Find the absolute maximum and minimum values of the function $f(x) = |x^2 - 2x - 3|$ over the interval $[0, 4]$.

Solution First defined the function.

```
f:=x->abs(x^2-2*x-3)
```

Find critical points.

```
with(Student[Calculus1]):  
Cpts:=CriticalPoints(f(x), x);
```

Let's put endpoints and critical points in $[0, 4]$ in a list.

```
lst:=[op(remove(c->c<0 or c>4, Cpts)),0,4];  
# The op function extracts elements from the list.  
# The remove function removes removes the elements of Cpts in [0,4]  
# The remove function use a Boolean-valued procedure as a condition.  
# A procedure can be considered as a function. This is why we use ->.
```

Evaluate the function at each point in the list.

```
fc:=f~(lst) # ~ is the element-wise operator.
```


Find the maximum and minimum.


```
fmax:=max(fc);
fmin:=min(fc);
```

You may use the following code to display at where the function reaches an extremum.

```
for c in lst do
    if f(c) = fmax then print(The maximum value of f(x) over [0, 4] is*f(c) = fmax);
    elif f(c)=fmin then print(The minimum value of f(x) over [0, 4] is*f(c) = fmin);
    end if;
end do;
```

Remark The above solution is still the method using Calculus but with the assistant of Maple. Indeed, Maple has the commands `maximize(f(x), x=a..b)` and `minimize(f(x), x=a..b)` which produce the maximum and minimum of a function $f(x)$ over an interval $[a, b]$.

 **Exercise 4.1** Find the absolute maximum and minimum values of the function $f(x) = x^3 - 3x + 1$ over the interval $[0, 2]$.

 **Exercise 4.2** Find the absolute maximum and minimum values of the function $f(x) = 2 \cos x - x - 1$ over the interval $[-2, 1]$.

4.2 The Mean Value Theorem

Rolle's Theorem states that if a function f is continuous on the closed interval $[a, b]$, differentiable on the open interval (a, b) and $f(a) = f(b)$, then there exists a point c in (a, b) such that $f'(c) = 0$. Using Maple, you can verify this theorem for a function f graphically using the following command `RollesTheorem(f(x), x = a..b)` which is supported by the subpackage `Student[Calculus1]`.

Example 4.2 Let $f(x) = (x-1)(x+1)(x-2)$. Graph the function $f(x)$ from -1 to 2 indicate the points between -1 and 2 where tangent line is horizontal. Find values $x = c$ in $(-1, 2)$ at where the tangent line is horizontal.

Solution Define the function.


```
restart:
f:=x->(x-1)*(x+1)*(x-2)
```

Now let's create the graph of the function and the horizontal tangent line.

```
with(Student[Calculus1]):
RollesTheorem((x-1)*(x+1)*(x-2), x = -1..2);
```

To find the values $x = c$. We solve the equation $f'(x) = 0$.

```
solve({D(f)(x)=0, x>-1, x<2}, x)
```



A demonstration of Rolle's theorem

A generalization of Rolle's theorem is the Mean Value Theorem.

If a function f is continuous on the closed interval $[a, b]$ and differentiable on the open interval (a, b) , then there exists a point c in (a, b) such that

$$f'(c) = \frac{f(b) - f(a)}{b - a}$$

or equivalently,

$$f(b) = f(a) + f'(c)(b - a).$$

The subpackage `Student[Calculus1]` also provides a command `MeanValueTheorem(f(x), x = a..b)` to illustrate the Mean Value Theorem.

Example 4.3 Let $f(x) = (x-1)(x+1)(x-2)$. Graph the function $f(x)$ from -2 to 3 indicate the points between -2 and 3 where tangent line is parallel to the secant line passing through $(-2, f(-2))$ and $(3, f(3))$. Find the values $x = c$ where the tangent line is parallel to the secant line.

Solution Define the function.

```
restart:
```

```
f:=x->(x-1)*(x+1)*(x-2)
```

Now let's create the graph of the function and the horizontal tangent line.

```
with(Student[Calculus1]):
```


```
MeanValueTheorem(f(x), x = -2..3);
```

Find the slope m_{sec} of the secant line.

```
a:= -2;
```

```
b:= 3;
```

```
msec:=(f(a)-f(b))/(a-b);
```

The Mean Value Theorem is illustrated for the function
 $(x-1)(x+1)(x-2) = (x-1)(x+1)(x-2)$.

A demonstration of Mean Value Theorem

Solve the equation $f'(x) = m_{sec}$ for c .

```
solve({D(f)(x)=msec, x>-2, x<3}, x)
```

Note that in Rolle's theorem and Mean Value Theorem, the continuity and differentiability conditions are crucial.

Example 4.4 Consider the floor function $f(x) = \begin{cases} 1 & x < 0 \\ 2 & x \geq 0 \end{cases}$. Use a graph to show that there is no tangent line that is parallel to the secant line through $(-1, 1)$ and $(1, 2)$.

Solution First define the function.

```
restart:
```

```
f := x->piecewise(x<0, 1, 2);
```

Now let's find the second line.

```
a := -1; # left endpoint
```

```
b := 1; # right endpoint
```

```
msec:=(f(b)-f(a))/(b-a); # slope of secant line
```

```
secline:=f(a) + msec*(x-a); # secant line
```


Plot the function and the secant line together.


```
plot([f(x), secline], x=a..b, discontinuity=true);
```


Now try the *MeanValueTheorem* command.

```
with( Student[Calculus1] );
```

```
MeanValueTheorem( f(x), a..b );
```

 **Exercise 4.3** Let $f(x) = \cos x$. Graph the function from $-\frac{2\pi}{3}$ to $\frac{4\pi}{3}$ and indicate the points in $[-\frac{2\pi}{3}, \frac{4\pi}{3}]$ where the derivative $f'(x)$ is 0. Find the tangent points where the slope of the tangent line is 0.

 **Exercise 4.4** Let $f(x) = x^4 - 3x^2 + 1$. Graph the function $f(x)$ from -1 to 2 indicate the points between -1 and 2 where tangent line is parallel to the secant line passing through $(-1, f(-1))$ and $(2, f(2))$. Find the tangent points where the tangent line is parallel to the secant line.

 **Exercise 4.5** Consider the floor function $f(x) = |x|$. Use a graph to show that there is no tangent line that is parallel to the secant line through $(-1, 1)$ and $(2, 2)$.

4.3 Derivatives and the Shape of a Graph

Given a function $f(x)$, we know that $f(x)$ is increasing on an interval if $f'(x) > 0$ on that interval and $f(x)$ is decreasing on an interval if $f'(x) < 0$ on that interval. By Mean Value theorem, $f(x)$ is a constant on an interval if $f'(x) = 0$ on that interval.

To determine over which interval a function $f(x)$ is increasing/decreasing, we need to find out the domain of $f'(x)$ and the critical points and then use test points to determine the sign of $f'(x)$.

Using first derivative, we can test local extrema at critical points.

Theorem 4.1 Suppose that c is a critical point of a continuous function f .

1. If f' changes from positive to negative at c , then f has a local maximum at c .
2. If f' changes from negative to positive at c , then f has a local minimum at c .
3. If f' does not change signs at c , then f has no local extremum at c .

Another method to determine local extremum is to use second derivative.

Theorem 4.2 Suppose f is a twice differentiable function on an interval I .

1. If $f'' > 0$ over I , then f is concave upward on I .
2. If $f'' < 0$ over I , then f is concave downward on I .

Base on the observation on concavity, we have the following second derivative test.

Theorem 4.3 Suppose that f'' is continuous near c .

1. If $f'(c) = 0$ and $f''(c) > 0$, then f has a local minimum at c .
2. If $f'(c) = 0$ and $f''(c) < 0$, then f has a local maximum at c .

Example 4.5 Find local extrema, the intervals on which $f(x) = \frac{x}{x^2+1}$ is increasing or decreasing and the intervals on which $f(x)$ is concave up or concave down.

Solution First we define the function and find the derivatives. $f:=x \rightarrow x/(x^2+1)$; $df:=D(f)$; $ddf:=D(D(f))$;

Now let's find critical points using *Student[Calculus1]* subpackage. `with(Student[Calculus1]): CriticalPoints(f(x),x)`

Find the intervals of monotonicity (using test points).

`df(t) # where t is a test point in an interval, you replace it by a number.`

Determine local extrema and find values.

`g(c)` # suppose `c` is a critical point.


Find critical points of the derivative function $g'(x)$ and determine intervals of concavity.

`CriticalPoints(df(x),x);`

`ddf(t);` # evaluate the second derivative at a test point `t`.

From the outputs, we know that f is decreasing on $(-\infty, -1) \cup (1, \infty)$ and increasing on $(-1, 1)$. It has a local minimum $f(-1) = -\frac{1}{2}$ and a local maximum $f(1) = \frac{1}{2}$.

The function is concave up on $(-\sqrt{3}, 0) \cup (\sqrt{3}, \infty)$ and concave down on $(-\infty, -\sqrt{3}) \cup (0, \sqrt{3})$.

 **Exercise 4.6** Let $f(x) = \sin x + \cos x$ be a function defined over the interval $[0, 2\pi]$. Find local extrema, the intervals on which f is increasing or decreasing and the intervals on which $f(x)$ is concave up or concave down.

4.4 Limits at Infinity and Asymptotes

Limits at infinity can provide information on the end behaviors of a curve such as horizontal and vertical asymptotes.

When the limit of a function is an infinite limits, we will get a vertical asymptotes.

The function f has a horizontal asymptote $y = b$ if $\lim_{x \rightarrow \infty} f(x) = b$ or $\lim_{x \rightarrow -\infty} f(x) = b$.

The function f has a horizontal asymptote $x = a$ if $\lim_{x \rightarrow a^+} f(x) = \infty$, $\lim_{x \rightarrow a^-} f(x) = \infty$, $\lim_{x \rightarrow a^+} f(x) = -\infty$ or $\lim_{x \rightarrow a^-} f(x) = -\infty$.

The function f has a slant asymptote $y = mx + b$ if $\lim_{x \rightarrow \infty} (f(x) - (mx + b)) = 0$ or $\lim_{x \rightarrow -\infty} (f(x) - (mx + b)) = 0$.

For rational functions, we have the following results. Suppose $p(x) = a_n x^n + \cdots + a_1 x + a_0$ and $q(x) = b_m x^m + \cdots + a_1 x + a_0$

$$\lim_{x \rightarrow \infty} \frac{p(x)}{q(x)} = \begin{cases} 0 & \text{if } n < m \\ \frac{a_n}{b_m} & \text{if } n = m \\ \pm\infty & \text{if } n > m \end{cases}$$

In the last case, the sign agrees with the sign of $\frac{a_n}{b_m}$.

The limit at the negative infinity of a rational function is similar.

In Maple, to find limits, we use the command `limit(f(x), x=a)` or `LimitTutor(f(x), x=a)` supported by `Student[Calculus1]`. To find asymptotes, you may use the command `Asymptotes(f(x), x)` which is again supported by `Student[Calculus1]`.

Example 4.6 Evaluate the following limits.

1. $\lim_{x \rightarrow \infty} \sqrt{x^2 + 1},$
2. $\lim_{x \rightarrow -\infty} \frac{x^3 - 1}{\sqrt{9x^6 - x}}$

3. $\lim_{x \rightarrow \infty} x \sin\left(\frac{1}{x}\right).$

Solution Use *LimitTutor* to find limits step-by-step

```
with(Student[Calculus1]);
LimitTutor(sqrt(x^2+1), x=infinity);
LimitTutor((x^3-1)/(9*x^6-x), x=-infinity);
LimitTutor(x*sin(1/x), x=infinity);
```

Example 4.7 Find asymptote of the function $f(x) = \frac{x^3+2x^2-3x-1}{x^2-1}$ and plot the graph of f and its asymptotes together with x in $[-5, 5]$ and y in $[-10, 10]$.

Solution Define the function first.

```
restart:
f:=x-> (x^3+2*x^2-3*x-1)/(x^2-1);
```

Find and plot asymptotes. Here, it's better to use *implicitplot* because vertical asymptotes are not functions of x .


```
with(Student[Calculus1]):
asym := Asymptotes(f(x), x);
with(plots):
asymplots := implicitplot(asym, x = -5 .. 5, y = -10 .. 10):
```


Plot the function f and use *display* to put the graph and asymptotes together.

```
Grphf:=plot(f(x), x=-5..5, y=-10..10, discontinuity=true):
display(asymplots, Grphf); # display is supported by the plots package.
```

 **Exercise 4.7** Evaluate the following limits.


1. $\lim_{x \rightarrow \infty} \sqrt{x} \sin\left(\frac{1}{x}\right),$
2. $\lim_{x \rightarrow -\infty} \frac{\sqrt{9x^6 + 1}}{x^3 + x}.$

 **Exercise 4.8** Find asymptote of the function $f(x) = \frac{2x^3-3x^2-5x+2}{x^2+1}$ and plot the graph of f and its asymptotes together with x in $[-5, 5]$ and y in $[-10, 10]$.

 **Exercise 4.9** Find asymptote of the function $f(x) = \frac{8 \sin x}{x^2+1} + \frac{x^2-x}{x^2-1}$ and plot the graph of f and its asymptotes together with x in $[-10, 10]$ and y in $[-20, 20]$.

4.5 Curve Sketching

To plot the graph of a function in maple, we simply use the command `plot(function, widows, options)`.



A graphs of rational functions with its asymptotes

On the other hand, we can sketch the graph using calculus, more precisely, monotonicity, concavity, vertical asymptotes, horizontal asymptotes, periods, symmetries, local extrema, intercepts etc.

Example 4.8 Sketch and plot the graph of the function $g(x) = \frac{2x+1}{x-2}$.

Solution First define the function

```
g:=x->(2*x+1)/(x-2)
```

Find the domain of the function.

```
solve((x-2)!=0, x)
```

Find the x -intercepts.

```
solve(g(x)=0, x)
```

Find the y -intercept

```
g(0)
```

Check whether $x = 2$, where the function is undefined, is a vertical asymptote.

```
limit(g(x), x=2, left);
```

```
limit(g(x), x=2, right);
```

Find horizontal asymptotes

```
limit(g(x), x=infinity);
```

```
limit(g(x), x=-infinity);
```

Then we find derivative functions.

```
dg:=D(g);    # first derivative
ddg:=D(dg); # second derivative
```

Find all critical points. This can be done by finding using the *Student[Calculus1]* subpackage.

```
with(Student[Calculus1]):
CriticalPoints(g(x),x);
```

Find the intervals of monotonicity (using test points).

```
dg(t) # where t is a test point in an interval, you replace it by a number.
```

Determine local extrema and find values.


```
g(c) # suppose c is a critical point.
```


Find critical points of the derivative function $g'(x)$ and determine intervals of concavity.

```
CriticalPoints(dg(x),x);
ddg(t);    # evaluate the second derivative at a test point $t$.
```

Sketch the graph using above obtained information and compare with Maple *plot* output.

```
plot(g(x), x=a..b) # Plot the function in the window [a, b].
```

 **Exercise 4.10** Sketch and plot the graph of the function $h(x) = \frac{x-1}{x+2}$.

 **Exercise 4.11** Sketch and plot the graph of the function $g(x) = x\sqrt{4-x}$.

4.6 Optimization Problems

To solve a optimization problem in Calculus, the key is to represent the quantity to be optimized by a function of other quantities and then find the extremum value using the extreme value theorem.

Example 4.9 Among rectangles with the same perimeter 16 centimeters, there is one that has the largest area. Find the dimension of that rectangle.

Solution Suppose the length is x and the width is y . The area is a function of x and y .

```
A:=(x, y)->x*y;
```

We know that the perimeter is 16 which provides a relation between x and y .


```
prm:=2x+2y=16;
```


Solve for y and plug it in to the area function.

```
wd:=solve(prm, y);
y:=unapply(wd, x); # define y as a function of x
Ar:=unapply(A(x,y(x)),x); # Translate the area function into a function of x.
```

To find the maximum area, you may use *maximize* or using the following commands. Note that $x > 0$.

```
Amx := max(Ar~(solve(D(Ar)(x) = 0, x))); # Find critical points, evaluate, and find the maximum
xvalue := solve(Ar(x) = Amx, x); # Find the length x such that the area is the largest.
yvalue := y(xvalue); # Find the width y such that the area is the largest.
```

 **Exercise 4.12** Find the rectangle with the minimal perimeter among rectangles with the area 36 square inches.

 **Exercise 4.13** Find the point on the parabola $y^2 = 2x$ that is closest to the point $(1, -4)$.

4.7 Antiderivatives

An antiderivative of a function f over an interval I is any function F such that $F'(x) = f(x)$ for all x in I .

Antiderivatives are closely related to integration. In Maple, you may use `int(f(x), x)` or `IntTutor(f(x), x)` supported by `Student[Calculus1]` to find an antiderivative of $f(x)$.

Example 4.10 Find an antiderivative of the function $f(x) = x^3 - \frac{1}{\sqrt{x}}$ by hand and by Maple respectively. Use `diff` to check your answer.

Solution To find an antiderivative by hand, rewrite the function using negative rational exponent and apply the formula

$$\frac{d}{dx} \left(\frac{1}{r+1} x^{r+1} \right) = x^r.$$

In this question, an antiderivative is $F(x) = \frac{1}{4}x^4 - 2\sqrt{x}$.

Define the function first.

```
restart:
f:=x->x^3-1/sqrt(x);
```




Find an antiderivative using *int* and using *IntTutor* for a step-by-step solution.

```
F:=int(f(x), x); # may define a function using unapply(int(f(x), x), x).
```

```
with(Student[Calculus1]):
TutorF:=IntTutor(f(x), x);
```

Verify using *diff*.

```
diff(F, x);
diff(TutorF, x);
```

-  **Exercise 4.14** Find an antiderivative of the function $f(x) = x^2 - 3 + 8\sin(x)$ by hand and by Maple respectively. Use `diff` to check your answer.
-  **Exercise 4.15** Find an antiderivative of the function $f(x) = \frac{x^3 - 2x + 1}{\sqrt{x}}$ by hand and by Maple respectively. Use `diff` to check your answer.
-  **Exercise 4.16** Find the most general antiderivative of the function $f(x) = \tan x \sec x + 1$ by hand and by Maple respectively. Use `diff` to check your answer.

Chapter 5 Integrals

5.1 Definite Integrals

In mathematics, the area of a region is measured by comparing it to squares of a fixed size. For example, the area of a dimension 1 by 3 rectangle equals the sum of three dimension 1 by 1 squares. For irregular shaped regions, the idea is similar. We slice the region into thin slices and estimate the area of each slice using a rectangle and then take the sum. In other words, the area of any region can be expressed as an infinite sum of rectangles of infinitesimal width. More precisely, to find the (signed) area under a curve $y = f(x)$ from a to b can be approximated using the Riemann Sum

$$R_N = \sum_{i=1}^N f(x_i^*)(x_i - x_{i-1}),$$

where x_i is a partition of the interval $[a, b]$ and x_i^* is a sample point in $[x_{i-1}, x_i]$.

If the limit R_N exists as the partition is getting finer and finer, that is $\max(x_i - x_{i-1}) \rightarrow 0$ and gives the same value for any possible choice of sample points, then we say the function f is integrable on $[a, b]$ and define the definite integral of f on $[a, b]$ as

$$\int_a^b f(x)dx = \lim_{\max(x_i - x_{i-1}) \rightarrow 0} R_n.$$

For an integrable function, we may approximate the definite integral by choosing left, lower, midpoint, right, upper sample points and take a regular subdivision of the interval as a partition, that is, take $x_i = a + i\Delta_x$, where $\Delta_x = \frac{b-a}{N}$

In Maple, we can use the following command to view and find Riemann sums approximately `RiemannSum(f(x), x = a..b, method, output, other options)`, where the `method` can be left, lower, midpoint, random, right, or upper, and the `output` can be value, sum, plot or animation. For details, you may use `?RiemannSum` to open the help page. Again, to run the command `RiemannSum` you need to first load the subpackage `Student[Calculus1]`.

Example 5.1 Find approximately the definite integral of the function $f(x) = x^3$ from 0 to 2.

Solution Load the package and apply the command.

```
with(Student[Calculus1]):  
RiemannSum(x^3, x = 0..2, method = right,  
  partition=50, output = plot,  
  boxoptions=[filled = [color = blue, transparency = .5]]);
```

Without using `RiemannSum`, you will also write your own program to approximate the integral. For example

Solution Another solution


restart:

```


f:=x->x^3;           # define the integrand
a:=0:                 # define the lower integral limit
b:=2:                 # define the right integral limit
Delta[x]:=(b-a)/N;    # define size of a regular subdivision
x[i]:=a+i*Delta[x];   # sample points using right endpoints
RS:=Sum(Delta[x]*f(x[i]), i=1..N); # find the sum
limit(RS, N=infinity); # evaluate the sum numerically.

```

Remark In the above code, you may use `sum` which is slightly more flexible than `Sum`. Indeed, `Sum(F, k=0..n) = sum(F, k=0..n, parametric)`. For details, see [Maple sum/details](#).

 **Exercise 5.1** Approximate the definite integral of the function $f(x) = \frac{x}{x+1}$ on the interval $[0, 1]$.

 **Exercise 5.2** Approximate the definite integral of the function $f(x) = \frac{x}{x+1}$ on the interval $[0, 3]$.

 **Exercise 5.3** Approximate the definite integral of the function $f(x) = \cos x$ on the interval $[0, \pi/2]$.

5.2 The Fundamental Theorem of Calculus

To evaluate definite integrals or find antiderivatives, we can use the Fundamental Theorems of Calculus.

Theorem 5.1 (Fundamental Theorem of Calculus I) If f is continuous on $[a, b]$, then the function g defined by

$$\frac{d}{dx} \int_a^x f(t) dt = f(x)$$

Theorem 5.2 (Fundamental Theorem of Calculus II) If f is continuous on $[a, b]$ and F is an antiderivative of f on $[a, b]$, then

$$\int_a^b f(t) dt = F(a) - F(b).$$

In maple, we can use `int(f(x), x=a..b)` or `IntTutor(f(x), x=a..b)` to find integrals.

Example 5.2 Find the derivative the function

$$g(x) = \int_0^x \sin(t) dt$$

and verify the Fundamental Theorem of Calculus I.

Solution Define the integrand.

```
f:=t->sin(t)
```

Define g as a function via the integral


```
g:=x->int(f(t), t=1..x)
```

Find the derivative function of g

```
dg:=x->diff(g(x), x)
```


Compare g' with f .

`evalb(dg(x)=f(x))` # `evalb` evaluates expression involving relational operators

 **Exercise 5.4** Use Maple to find the derivative of the following function

$$g(x) = \int_0^{\tan x} \sqrt{t} dt$$

and verify your answer using the Fundamental Theorem of Calculus I (find the derivative by hand).

 **Exercise 5.5** Find $g(x) = \int_0^x \sqrt{1-t^2} dt$ and verify that $g'(x) = \sqrt{1-x^2}$ and $\int_0^1 \sqrt{1-t^2} dt = g(1) - g(0)$.

5.3 Indefinite Integrals

When calculating a definite integral, we first find an antiderivative and then apply the Fundamental Theorem of Calculus. To be convenient, we use the $\int f(x)dx$ to denote the most general antiderivative of f and call it the indefinite integral of f .

In Maple, you may use `int`, `IntTutor` or `Int` to find an integral, where `Int` should be used with `value` if you want to see the evaluated integral. Again, `IntTutor` is supported by `Student[Calculus1]`.

Example 5.3 Find the indefinite integral

$$\int (x^3 - \sin x) dx.$$


Show the steps.

Solution Define the integrand.

```
restart:
f:=x->x^3-sin(x);
```


Find the indefinite integral step-by-step using *IntTutor*.

```
with(Student[Calculus1]):
IntTutor(f(x), x);
```

 **Exercise 5.6** Find the indefinite integral

$$\int (\sqrt{x^3} - \sec^2 x) dx.$$

Show the steps.

 **Exercise 5.7** Find the indefinite integral

$$\int \left(\frac{x^2 + 1}{\sqrt{x}} - \tan x \sec x \right) dx.$$

Show the steps.

5.4 The Substitution Rule

Among techniques of integration, one basic rule is the substitution rule.

If $g'(x)$ is continuous on $[a, b]$ and f is continuous on the range of $u = g(x)$, then

$$\int_a^b f(g(x))g'(x)dx = \int_{g(a)}^{g(b)} f(u)du.$$

In Maple, we can use the command `Change(Int(f(x), x=a..b), u=g(x), new variables)`, supported by the package `IntegrationTools`, to change a variable, where the 3rd argument is optional and required only if the number of new symbols in the substitution is not equal to the number of old variables. Note that in the first argument, you may keep the symbols `a` and `b` or omit `=a..b`.

Example 5.4 Use substitution to find the following integral.

$$\int_1^2 x\sqrt{x^2+1}dx.$$

Solution When using substitution to find a definite integral, it's better to find the indefinite integral first and then use the Fundamental Theorem of Calculus to find the value..

In this question, we may use the substitution $u = x^2 + 1$.

Let's use Maple to see how variables will be changed.

First define the function.

```
f:=x->x*sqrt(x^2+1)
```

Load the package *IntegrationTools*

```
with(IntegrationTools)
```

Find the indefinite integral by the substitution $u = x^2 + 1$.

```
F:=Int(f(x), x);
```

```
FInt:=Change(F, u=x^2+1);
```

Evaluate integral in u and substitute back to x .

```
G:=unapply(value(FInt), u);
```


```
FF:=unapply(G(x^2+1), x);
```

Apply the Fundamental Theorem of Calculus I to find the definite integral.

```
Ans:=FF(2)-FF(1)
```

Now let's check the answer using the `int(f(x), x=a..b)` command.

```
int(f(x), x=1..2)
```

 **Exercise 5.8** Find the definite integral $\int_0^4 \frac{\sin(\sqrt{x})}{\sqrt{x}} dx$ using substitution.

 **Exercise 5.9** Find the definite integral $\int_1^2 \frac{2x}{\sqrt{x^2+1}} dx$ using substitution.

Chapter 6 Application of Integrals I

6.1 Areas Between Curves

By the definition of integral, we see that the area between two curves, is the difference of areas “under” each curve. Here “under” is relative, it depends on how you slice the area.

If the slices are vertical, then the area between two curves $y = T(x)$ and $y = B(x)$ from a to b is given by

$$\int_a^b |T(x) - B(x)| dx.$$

If the slices are horizontal, then the word “under” should be understood horizontally and the area between two curves $x = R(y)$ and $x = L(y)$ from c to d may be calculated by

$$\int_c^d |R(y) - L(y)| dy.$$

Before you setup the interval, it’s better to plot the curves first.

Example 6.1 Plot the curves $y = x^2$ and $y = 2x - x^2$, and find the enclosed area

Solution Define the functions.

```
y[1]:=x->x^2:
```

```
y[2]:=x->2*x-x^2:
```

Plot the curves and shade the enclosed area

```
Curves:=plot([y[1](x), y[2](x)], x=-1..2, color=[red, blue])
```

From the graph, we see the curves intersect at two points. Let’s find the x -coordinates of the intersection points which are the integral limits.

```
pts:=solve(y[1](x)=y[2](x), x)
```

Determine the upper and lower limits of the integral.

```
a:=min(pts):
```

```
b:=max(pts):
```


Setup the integral and evaluate

```
Area:=Int(abs(y[1](x)-y[2](x)), x=0..1);
```

```
ValArea:=value(Area);
```

Example 6.2 Find the area enclosed by the curves $y = x - 1$ and $y^2 = 2x + 6$.

Solution Note the second curve does not define a function of x . So let’s just define the curves as equations.




A plot of two curves

```
C[1] := y = x + 1;
C[2] := y^2 = 2x + 6;
```

Plot the two curves use *implicitplot* which is supported by *plot*.

```
with(plots):
implicitplot([C[1], C[2]], x = -5..5, y = -5..5, color = [red, blue]);
```



An implicit plot of two curves

From the graph, we see that it's better to integrate along y -axis. So we need to solve for x from the equations and view it as a function of y .




```
x[1]:=unapply(solve(C[1], x), y);  
x[2]:=unapply(solve(C[2], x), y);
```

Find the y -coordinate of the intersection points and setup integral limits.

```
ycord:=solve(x[1](y)=x[2](y), y);  
a:=min(ycord):  
b:=max(ycord):
```

Setup the integral with respect to y and evaluate.

```
Area:=Int(abs(x[1](y)-x[2](y)), y=a..b);  
ValArea:=value(Area);
```

-  **Exercise 6.1** Find the area of the region bounded by $y = \sin x$, $y = \cos x$, $x = 0$ and $x = \pi/2$.
-  **Exercise 6.2** Find the area between the curve $x = y^2 - 4y$ and $x = 2y - y^2$.
-  **Exercise 6.3** Find the area of the ellipse $\frac{x^2}{4} + \frac{y^2}{9} = 1$.

Part 2 - Calculus II

Chapter 7 Applications of Integrals II

7.1 Volume of Revolution

In terms of definite integrals, the volume of a solid obtained by rotating a region about the x -axis can be calculated by

$$\int_a^b \pi(r_1(x)^2 - r_2(x)^2) dx \quad \text{disk/washer method,}$$

or

$$\int_a^b 2\pi r(y)h(y) dy \quad \text{shell method method,}$$

where $r_1(x)$, $r_2(x)$ and $r(x)$ represents the radius and $h(x)$ represents the height of a cylindrical shell.

In practice, it's better to recognize the shape of a cross section, find the volume of a slice of the solid and then set up the integral.

In the following, you will see some tools/commands from Maple which are very helpful to calculate the volume of a solid.

In Maple, the following command, supported by the package `Student[Calculus1]`, can be used to get the graph, the integral and the volume of the solid obtained by rotation the region bounded by $f(x)$, $g(x)$, $x = a$ and $x = b$.

`VolumeOfRevolution(f(x), g(x), x = a..b, opts)`

To learn what options does the command `VolumeOfRevolution` have, you may type

`?VolumeOfRevolution`

in the Math mode and hit enter. You will see the help page.

Student[Calculus1] VolumeOfRevolution find the volume of revolution of a curve	
Calling Sequence	Notes
Parameters	Examples
Description	Compatibility
Calling Sequence <code>VolumeOfRevolution(f(x), x = a..b, opts)</code> <code>VolumeOfRevolution(f(x), g(x), x = a..b, opts)</code> <code>VolumeOfRevolution(f(x), a..b, opts)</code> <code>VolumeOfRevolution(f(x), g(x), a..b, opts)</code>	
Parameters	
<code>f(x), g(x)</code>	- algebraic expressions in variable <code>x</code>
<code>x</code>	- name; specify the independent variable
<code>a, b</code>	- algebraic expressions; specify the endpoints of the curve
<code>opts</code>	- equation(s) of the form <code>option=value</code> where <code>option</code> is one of <code>axis</code> , <code>distancefromaxis</code> , <code>functionoptions</code> , <code>functionoptions</code> , <code>lineoptions</code> , <code>method</code> , <code>numpoints</code> , <code>output</code> , <code>partition</code> , <code>regionoptions</code> , <code>revolutionpoints</code> , <code>showfunction</code> , <code>showrotationline</code> , <code>showaxis</code> , <code>showvolume</code> , <code>sumvolumeoptions</code> , <code>volumeoptions</code> , <code>volumeoptions</code> , <code>border</code> , or <code>Student[plot] options</code> ; specify output options

Maple help page for `VolumeOfRevolution` command

Example 7.1 Show the solid obtained by rotating the region bounded by $y = x^2$ and $y = x$ about y -axis. Set up an integral for the volume. Find the volume.

Solution

```
#Load the package
```

```
with(Student[Calculus1])
```

```
#Show the solid
```

```
VolumeOfRevolution(x^2, x, x = 0 .. 1, axis = vertical, output = plot)
```


```
#Set up an integral
```

```
VolumeOfRevolution(x^2, x, x = 0 .. 1, axis = vertical, output = integral)
```

```
#Find the volume
```

```
VolumeOfRevolution(x^2, x, x = 0 .. 1, axis = vertical, output = value)
```

The outputs in Maple can be seen in the following picture





Volume of Revolution Example

Remark 1. If you change the function to `VolumeOfRevolutionTutor`, you will see an interactive popup windows which does exactly the same thing.

2. If the rotation axis is not an axis of the coordinate system, you need add the option `distancefromaxis = numeric` into the function. For example, if in the above example, the rotation is about $y = -2$, then the Maple command should be the following

```
VolumeOfRevolution(x^2, x, x = 0 .. 1, axis = vertical, distancefromaxis = -2, output = integ
```

 **Exercise 7.1** Find the volume of the solid obtained by rotating the region bounded by $y = x^3$, $x = 0$, $y = 1$ about y -axis

 **Exercise 7.2** Find the volume of the solid obtained by rotating the region bounded by $y = x^3$, $y = 0$, $x = 1$ about (a) $y = 0$, and (b) $x = 2$.

Chapter 8 Calculus of Inverse Functions

8.1 Inverse Functions

Maple package `Student[Calculus1]` provides the following command

```
InversePlot(f(x), x = a..b);
```

which graphs the original function $f(x)$ and the inverse function $f^{-1}(x)$ together over the interval $[a, b]$.

You will see clearly that the graphs of a function and its inverse are symmetric with respect to the line $y = x$.


Example 8.1

1. Graph the function $f(x) = x^3 - 2$, its inverse function, and the line $y = x$ over the interval $[-2, 2]$.
2. Find the inverse function.

Solution One way to plot the function and its inverse together is to use the following command which is supported by the package `Student[Calculus1]`.

```
InversePlot(x^3-3, x = -2..2)
```

Here is the output in Maple



Graph of a pair of functions inverse to each other

Another way to plot the function f and its inverse g together uses the `plot` function.

```
plot([f(x), g(x), x], x = -2 .. 4, y = -5 .. 5, color = [red, black, blue])
```

To find the inverse function, we replace $f(x)$ by y , then switch x and y , and solve for y . In Maple, you may use the command `solve(equation/inequality, variable)` to solve an equation or an inequality (even system of equations/inequalities).

In this example, we may find the inverse function by type in the following command. Note I have switch x and y .

```
solve(x=y^3, y)
```

To find the derivative of the inverse function of a function f at a given point $x = a$, we may apply the formula

$$(f^{-1})'(a) = \frac{1}{f'(f^{-1}(a))}.$$

In Maple, we may use the following commands to calculate the value of the derivative function.

- Calculate the derivative of the function f .

```
diff(f(x), x)
```

- Find $f^{-1}(a)$ which is the solution of the equation $f(x) = a$.

```
solve(f(x)=a, x)
```

- Plug in the formula to evaluate.

```
eval(subs(x=f^{-1}(a), 1/f'(x)))
```

Example 8.2 Find $(f^{-1})'(0)$, where $f(x) = \cos(x)$ and $0 \leq x \leq \pi$.

Solution Find the derivative of f

```
diff(cos(x), x)
```


Find the value of $f^{-1}(0)$

```
solve(cos(x)=0, x)
```

Apply the formula

```
eval(subs(x=Pi/2, -1/sin(x)))
```

Using Maple, we find $(f^{-1})'(0) = -1$.

 **Exercise 8.1** 1. Graph the function $f(x) = 3 + 2 \sin x$, its inverse function, and the line $y = x$ over the interval $[-2, 2]$.

2. Find the value $(f^{-1})'(5)$.

8.2 Logarithmic and Exponential Functions

8.2.1 Basic properties and graphs

The natural logarithmic function $y = \ln(x)$ is defined by $\ln(x) = \int_1^x \frac{1}{t} dt$.

The natural exponential function $y = e^x$ is defined as the inverse function of $y = \ln(x)$.

From the definition, we have very important identities

$$\ln(e^x) = x \quad \text{and} \quad e^{\ln x} = x.$$

Using those two identities, we may define general exponential functions and general logarithmic function, and deduce the Law of Logarithms and Law of Exponents.



- For any positive number $b \neq 1$, we have $b^x = (e^{\ln b})^x = e^{x \ln b}$.
- For any positive number $b \neq 1$, we define $y = \log_b x$ to be the inverse function of $y = b^x$
- By solving $x = b^y$ for y , we find that $\log_b x = \frac{\ln x}{\ln b}$. This identity is called the change of base property.

How do graphs of logarithmic functions and exponential functions look like?

Example 8.3 Graph the following functions together.

$$y = \ln x, \quad y = e^x, \quad y = 2^x, \quad y = \log_2 x, y = x.$$


Solution In Maple, the logarithm $\log_b x$ is given by `log[b](x)`. When $b = e$, you simply use `ln(x)` for $\ln x$. When $b = 10$, you may also use `log(x)` or `log10(x)` for $\log_{10} x$.

The exponent b^x is given by `b^x` in Maple. When $b = e$, you may also use `exp(x)` to represent e^x .


To graph the functions together with different colors, we use the following command

```
plot([ln(x), exp(x), 2^x, log[2](x), x], x=-5..5, color=[blue, green, purple, yellow, red])
```

Here is the output




Logarithmic and exponential functions

 **Exercise 8.2** Graph the following functions together.

$$y = \log_3 x, \quad y = 3^x, \quad y = (1/3)^x, \quad y = \log_{1/3} x.$$


Find the pairs that are symmetric to each other with respect to a certain line.

 **Exercise 8.3** Graph the following functions together.

$$y = 0.5^x, \quad y = 2^x, \quad y = 5^x.$$

Describe the monotonicity (increasing/decreasing) of the functions?

Fix an input x . Describe how y -values change when bases changes from small number to bigger number?

 **Exercise 8.4** Graph the following functions together.

$$y = \log_{0.5} x, \quad y = \log_2 x, \quad y = \log_5 x.$$




Describe the monotonicity (increasing/decreasing) of the functions?

Fix an input x . Describe how y -values change when bases change from small number to bigger number?

8.2.2 Differentiation and integration of logarithmic and exponential functions

In Maple, one way to do differentiation and integration is to use the **Calculus Palette** on the left side.



Calculus Palette in Maple

The other way is to use the commands `diff(f(x), x)`, `int(f(x), x)`, and `int(f(x), x=a..b)`. Supported by the `Student[Calculus1]` package, Maple also provides the tutor commands `DiffTutor()` and `IntTutor()` which can show step-by-step solution of differentiation and integration.

Note you may also access tutor commands from the **Start** page (click the home button in the toolbar and look for Calculus).

Example 8.4 Find y' , where $y = \ln(x^3 + 5x + 1)$.

Solution Using *diff*:

```
diff(ln(x^3+5*x+1), x)
```


We get

$$y' = \frac{3x^2 + 5}{x^3 + 5x + 1}.$$

Type in (assume that $\text{with}(\text{Student}[\text{Calculus1}])$ was run)

`DiffTutor(ln(x^3+5*x+1), x)`

and hit enter you will see



DiffTutor Example

By click *Next Step* or *All Steps* you will see detailed solution with rules used.

Example 8.5 Evaluate the integral

$$\int \frac{e^x - 1}{e^x + 1} dx.$$

Solution Using *int*:

`int((exp(x)-1)/(exp(x)+1), x)`

We get

$$\int \frac{e^x - 1}{e^x + 1} dx = 2 \ln(e^x + 1) - x + C.$$

Type in (assume that $\text{with}(\text{Student}[\text{Calculus1}])$ was run)

`IntTutor((exp(x)-1)/(exp(x)+1), x)`

and hit enter you will see

By click *Next Step* or *All Steps* you will see detailed solution with rules used.

Exercise 8.5 Find the derivative $\frac{dy}{dx}$, where $y = \ln |\cos x|$

Exercise 8.6 Find the derivative $\frac{dy}{dx}$, where $y = x^{\cos x}$

Exercise 8.7 Evaluate the integral

$$\int \frac{(e^{4x} + e^{2x})}{e^{3x}} dx$$

Exercise 8.8 Evaluate the integral

$$\int 2^{3x} dx$$

IntTutor Example

8.3 Solve differential equations

In Maple, you may solve the equation $y'(x) = ky(x) + c$ (which is called an ODE) using the command `dsolve({ics, eq})`, where `ics` stands for initial condition $y(0) = c$ and `eq` stands for the differential equation. Without the `ics`, `dsolve` will provide a general solution.

Example 8.6 Find the function $f(x)$ which satisfies the differential equation $f'(x) = kf(x)$ with $f(0) = 5$ and $f(2) = 3$.

Solution Use the following command

```
dsolve({f(0)=5, f'(x)=k f(x)})
```

we get $f(x) = 5e^{kx}$.


To find k , we solve the equation $3 = 5e^{2k}$ by

```
solve(3=5*e^(2*k), k)
```

which shows that $k = \frac{\ln 3 - \ln 5}{2} \approx -0.255$. Here we use `evalf(%)` (%) represents the previous result) to get the approximation.

So the function f is given by

$$f(x) = 5e^{\frac{x(\ln 3 - \ln 5)}{2}} \approx 5e^{-0.255x}.$$

 **Exercise 8.9** Find the function y which satisfies the differential equation $y'(x) = ky(x)$ with $y(0) = 2$ and $y(5) = 11$.

8.4 Inverse Trigonometric Functions

8.4.1 Domains and Ranges

To define the inverse function, the original function must be a one-to-one function. For a trigonometric function, we have to restrict the function over a specific domain to ensure that the function is one-to-one. For simplicity, we pick domains near the origin for trigonometric functions. To be more precise, we consider the following trigonometric functions:

$$y = \sin x, \quad -\pi/2 \leq x \leq \pi/2 \quad \text{and} \quad -1 \leq y \leq 1;$$

$$y = \cos x, \quad 0 \leq x \leq \pi \quad \text{and} \quad -1 \leq y \leq 1;$$

$$y = \tan x, \quad -\pi/2 < x < \pi/2 \quad \text{and} \quad -\infty < y < \infty.$$

Their inverse functions are

$$y = \arcsin x, \quad -\pi/2 \leq y \leq \pi/2 \quad \text{and} \quad -1 \leq x \leq 1;$$

$$y = \arccos x, \quad 0 \leq y \leq \pi \quad \text{and} \quad -1 \leq x \leq 1;$$

$$y = \arctan x, \quad -\pi/2 < y < \pi/2 \quad \text{and} \quad -\infty < x < \infty.$$


To see the graphs of the functions, we may use the `plot(f(x), x = a..b, opts)` command. Or, to graph f , f^{-1} and $y = x$ together, we may also use the command `InversePlot(f(x), x = a..b, opts)` supported by the package `Student[Calculus1]`.

Example 8.7 Graph the following functions together.

$$y = \tan x, \quad y = \arctan x, \quad y = x.$$

Solution `#load the package "Student[Calculus1]". with(Student[Calculus1]) #plot the functions`
`InversePlot(tan(x), x=-Pi/2..Pi/2)`

Here is the output

 **Exercise 8.10** Graph the following functions together over an appropriate domain.

$$y = \cot x, \quad y = \operatorname{arccot} x, \quad y = x.$$

What's the domain and range of $y = \operatorname{arccot} x$?


8.4.2 Differentiation and integration of inverse trigonometric functions

In the section {#Differentiation and integration of logarithmic and exponential functions}, we learned to how use Maple to learn differentiation and integration.

Now let find derivatives and integrals of some inverse trigonometric functions.

Example 8.8 Find y' , where $y = \operatorname{arccot} x$.

Solution Surely, we may use *diff* or *DiffTutor* to find the derivative.



Tangent and arctangent functions

Here let's me introduce to you another command *implicitdiff*.

We've learned that (see {#Inverse Functions}) to find the inverse function, we switch x and y and then solve for y . When finding the derivative, we don't have to solve for y instead, we want y' which is implicitly defined by an equation. In this case, we have $x = \operatorname{arccot} y$.

Enter the following commands in Maple, you will find $y' = -\frac{1}{x^2+1}$.

```
implicitdiff(x=cot(y), y, x)
subs(cot(y) = x, %)
```

where % is a ditto operator that allows you to refer to a previously computed result in Maple.

Exercise 8.11 Find the derivative of $y = \operatorname{arcsec} x$

Exercise 8.12 Find the derivative of $y = \operatorname{arccsc} x$

For integrals of inverse trigonometric function, you may need the method of integration by parts.

Use **DiffTutor** to find antiderivatives of inverse trigonometric functions.

Exercise 8.13 Evaluate the integral

$$\int \arcsin x dx$$

Exercise 8.14 Evaluate the integral

$$\int \arctan x dx$$

Exercise 8.15 Evaluate the integral

$$\int \sec x dx$$

8.5 L'Hospital's Rule

In Maple, supported by the package, `Student[Calculus1]`, the command `LimitTutor` can show step-by-step solutions of evaluating limits.

Example 8.9 Evaluate the limit

$$\lim_{x \rightarrow \infty} (1+x)^{1/\ln(x)}$$

Solution # load the package `Student[Calculus1]`. `with(Student[Calculus1])` #Find the limit step-by-step using `LimitTutor` `LimitTutor((1+x)^(1/ln(x)), x = infinity)`

 **Exercise 8.16** Estimate the limit

$$\lim_{x \rightarrow 1} \frac{x^2 - 2x + 1}{x^2 - x}$$

by graphing and verify your estimation.

 **Exercise 8.17** Evaluate the limit

$$\lim_{x \rightarrow \infty} x - \ln(x).$$

 **Exercise 8.18** Evaluate the limit

$$\lim_{x \rightarrow \infty} x \tan\left(\frac{1}{x}\right).$$

Chapter 9 Techniques of Integration

9.1 Integrations of trigonometric functions

When evaluating integrations of trigonometric functions, one idea is to reduce the total degree (power) of trigonometric functions using trigonometric identities.

In Maple, you may use the command `combine` to rewrite the expression.

Example 9.1 Rewrite $\cos^4 x$ into an expression with single terms and evaluate the integral $\int \cos^4 x dx$.

Solution

```
# combine terms into a single term
combine((cos(x))^4)
# use DiffTutor to evaluate the integral of the resulting function.
with(Student[Calculus1])
IntTutor(%, x)
```

You may compare the above solution with the solution given by `IntTutor((cos(x))^4, x)`.

 **Exercise 9.1** Evaluate the integral

$$\int \tan^5 x dx$$

 **Exercise 9.2** Evaluate the integral

$$\int \sin 5x \sin^2 x dx$$

9.2 Trigonometric Substitution

Surely, you may learn some trigonometric substitution tricks using `IntTutor`.

Here I want to introduce another useful command which when integrating functions, we may need to complete a square and then do a substitution. In Maple, we can complete squares using the command `CompleteSquare(f, x)` which supported by the package `Student[Precalculus]`.

Example 9.2 Evaluate the integral

$$\int \frac{1}{x^2 + x + 1} dx.$$

Solution

We first complete the square for the denominator.

```
#load package Student[Precalculus]
with(Student[Precalculus])
#Complete square for the denominator
CompleteSquare(x^2+x+1, x)
```

Now you may try *DiffTutor* and/or evaluate it by hand.

```
#load package Student[Calculus1]
with(Student[Calculus1])
DiffTutor(1/%, x)
```

 **Exercise 9.3** Evaluate the integral

$$\int \sqrt{3 + 2x - x^2} dx$$

 **Exercise 9.4** Evaluate the integral

$$\int_0^{\pi/2} \frac{\cos t}{\sqrt{1 + \sin^2 t}} dt$$

9.3 Integrations of Rational Functions by Partial Fractions


In Maple, we can factor a polynomial using the command `factor(polynomial)` or find partial fraction decomposition using `convert(function, parfrac)`.

Example 9.3 Find the sum of partial fractions for the rational function

$$f(x) = \frac{x^3 + 4x + 3}{x^4 + 5x^2 + 4}$$

Solution This can be done easily in Maple:


```
# use the command convert
convert((x^3+4*x+3)/(x^4+5*x^2+4), parfrac)
```

 **Exercise 9.5** Find the sum of partial fractions for the rational function


$$f(x) = \frac{x^4}{(x^2 - x + 1)(x^2 + 2)^2}.$$

 **Exercise 9.6** Find the sum of partial fraction and evaluate the integral

$$\int \frac{2}{3x^2 + 2x - 1} dx$$

 **Exercise 9.7** Find the sum of partial fraction and evaluate the integral

$$\int \frac{x^3 + 6x - 2}{x^4 + 6x^2} dx$$

 **Exercise 9.8** Find the sum of partial fraction and evaluate the integral

$$\int \frac{\sin x}{\cos^2 x - 3 \cos x} dx$$

Chapter 10 Further Applications of Integration

10.1 Arc Lengths and Areas of Surfaces of Revolutions

- Arc length:

The length L of an arc: $y = f(x)$, $a \leq x \leq b$ is

$$L = \int_a^b \sqrt{1 + \left(\frac{dy}{dx}\right)^2} dx = \int_{f(a)}^{f(b)} \sqrt{1 + \left(\frac{dy}{dx}\right)^2} dy.$$

- Surface area of a revolution

The area S of the surface rotating an arc: $y = f(x)$, $a \leq x \leq b$ about the x -axis is

$$S = 2\pi \int r ds = 2\pi \int y ds,$$

and about the y -axis is

$$S = 2\pi \int r ds = 2\pi \int x ds,$$

where

$$ds = \sqrt{1 + \left(\frac{dy}{dx}\right)^2} dx = \sqrt{1 + \left(\frac{dy}{dx}\right)^2} dy.$$

The integral limits depend on whether you use dx or dy in the integral.

In Maple, the package `Student[Calculus1]` provides commands to investigate arc length and surface area of revolutions: `ArcLength(f(x), x = a..b, opts)}` `SurfaceOfRevolution(f(x), x = a..b, opts)}`

Example 10.1 Set up an integral and evaluate the integral for the length of the curve defined by

$$f(x) = \sqrt{x}, \quad 1 \leq x \leq 4.$$

Plot $f(x)$ together with the arc length function in the same coordinate system.

Solution

```
# Load the package
```

```
with(Student[Calculus1])
```

```
# Set up an integral
```

```
ArcLength(sqrt(x), x=1..4, output=integral)
```

```
# Evaluate the integral
```

```
ArcLength(sqrt(x), x=1..4)
```

Plot the function and the arc length function

```
ArcLength(sqrt(x), x=1..4, output=plot)
```

Example 10.2 Set up an integral and evaluate the integral for the area of the surface obtained by rotating the curve defined by

$$f(x) = \sqrt{x}, \quad 1 \leq x \leq 4$$

about the y -axis. Plot the surface of the revolution.

Solution

Load the package (skip if the package was already loaded)

```
with(Student[Calculus1])
```

Plot the surface

```
SurfaceOfRevolution(sqrt(x), x=1..4, output=plot, axis=vertical)
```

Set up an integral

```
SurfaceOfRevolution(sqrt(x), x=1..4, output=integral, axis=vertical)
```

Evaluate the integral

```
SurfaceOfRevolution(sqrt(x), x=1..4, axis=vertical)
```

 **Exercise 10.1** Set up an integral and evaluate the integral for the length of the arc defined by

$$f(x) = \ln x, \quad 1 \leq x \leq 2.$$

Plot $f(x)$ together with the arc length function in the same coordinate system.

 **Exercise 10.2** Plot the surface obtained by rotating the curve defined by

$$f(x) = \frac{\cos x}{x}, \quad 0 \leq x \leq 4\pi$$

about the y -axis. Set up an integral for the area of the surface.

 **Exercise 10.3** Find the area of the surface obtained by rotating the curve defined by

$$f(x) = \sqrt{1 + x^2}, \quad 0 \leq x \leq 3.$$

Chapter 11 Infinite Sequences and Series

11.1 Introduction to Sequences and Series

A sequence is a list of numbers in a definite order (indexed by integers). A series may be considered as the limit of the sequence of partial sums.

When the sequence is explicitly defined by an mathematical expression $a_n = f(n)$, Maple has the following command to list numbers of the sequence `seq(f, i=m..n, step)`.

Example 11.1 Find the first 10 terms of the sequence $\{\frac{1}{n(n+1)}\}_{n=1}^{\infty}$. Determine whether the sequence $\{\frac{1}{n(n+1)}\}$ is convergent or divergent.

Solution

```
# using seq
```

```
seq(1/(n*(n+1)), n=1..10)
```

The sequence converges to 0.

For a series $\sum a_n$, normally it is not easy to find explicit expression for the partial sum $s_n = \sum_{k=1}^n a_k$.

However, if sequence is defined by an mathematical expression $a_n = f(n)$, we may find values of partial sums recursively use a `for/from loop` statement in Maple:

```
for *counter* from *initial* by *increment* to *final* do
    statement_sequence;
end do;
```

Example 11.2 Find the first 20 partial sums $s_k = \sum_{n=1}^k a_n$ of the infinite series

$$\sum_{n=0}^{\infty} \frac{1}{2^n} = 1 + \frac{1}{2} + \frac{1}{4} + \frac{1}{8} + \cdots.$$

Determine whether the series $\sum_{n=0}^{\infty} \frac{1}{2^n}$ is convergent or divergent.

Solution

```
# Set up s when n=0
```

```
s:=1
```

```
# Find 10 terms using `for/from loop`
```

```
for n from 1 to 10 do
    s:=s+1/(2^n);
end do;
```


The series converges to 2.

Of course, we may also use `for/from loop` to list numbers of a sequence.

Solution Second solution to example 11.1.

```
# using `for/from loop`

for n from 1 to 10 do
    1/(n*(n+1));
end do;
```

When the sequence is defined by a recurrence formula like the Fibonacci sequence, we will need to Maple how to interpret the formula. For that purpose, we use a procedure, which encloses a sequence of statements between `proc(...)` and `end proc`, to define the formula in Maple.

For example, the following is a procedure that defines a function $a(x) = \sqrt{x} - \frac{1}{\sqrt{x}}$:

```
a:=proc(x) sqrt(x)-1/sqrt{x}; end proc;
```

To structure codes in a procedure, you may use **Code Edit Region** which can be find in the **Insert** menu. To execute codes within this region, click **Execute Code** from the **Edit** menu, or use the shortcut command **Ctrl+E**.


Example 11.3 The Fibonacci sequence is defined by $fib(0) = 0$, $fib(1) = 1$ and $fib(n) = fib(n - 1) + fib(n - 2)$. Find the first 20 Fibonacci numbers.

Solution We first define a function $fib(n)$ which returns the n -th Fibonacci number.

```
fib := proc (n::nonnegint)
    if 2 <= n then
        return fib(n-1)+fib(n-2);
    else
        return n;
    end if;
end proc
```


Now we can use either `seq()` or `for/from loop`.

```
seq(fib(n), n=0..19)
```

 **Exercise 11.1** Find the first 20 terms of the sequence

$$\left\{ \sin \frac{\pi}{n} \right\}_{n=1}^{\infty}.$$

Determine whether the sequence $\left\{ \sin \frac{\pi}{n} \right\}$ is convergent or divergent.

 **Exercise 11.2** Find the first 20 partial sums $s_k = \sum_{n=1}^k a_n$ of the infinite series

$$\sum_{n=0}^{\infty} \frac{1}{n} = 1 + \frac{1}{2} + \frac{1}{3} + \frac{1}{4} + \cdots.$$

Determine whether the series $\sum_{n=0}^{\infty} \frac{1}{n}$ is convergent or divergent.

 **Exercise 11.3** Find the 20th to 30th Fibonacci numbers.

11.2 Power Series

A power series is a series with a variable x :

$$\sum_{n=0}^{\infty} c_n x^n = c_0 + c_1 x + c_2 x^2 + c_3 x^3 + \cdots.$$

More generally, a series of the form

$$\sum_{n=0}^{\infty} c_n (x-a)^n = c_0 + c_1 (x-a) + c_2 (x-a)^2 + c_3 (x-a)^3 + \cdots \quad (11.1)$$

is called a power series at a .

We call a positive number R the radius of convergence of the power series (11.1) if the power series converges whenever $|x-a| < R$ and diverges whenever $|x-a| > R$.

If a function f has a power series representation, i.e.

$$f(x) = \sum_{n=0}^{\infty} c_n (x-a)^n, \quad |x-a| < R,$$

then its coefficients are given by $c_n = \frac{f^{(n)}(a)}{n!}$.

Example 11.4 Find the interval of convergence of the power series

$$\sum_{n=1}^{\infty} \frac{(-2)^n x^n}{n^3}.$$

Solution

```
# Find the abs(a_{n+1}/a_n)
```

```
q:=abs((-2)^(n+1)(n+1)^3/(-2)^(n+1)(n+1)^3);
```

```
# Find the limit of q
```

```
r:=limit(simplify(q), n=infinity)
```

```
# Find the interval of convergence
```

```
solve(abs(x)<1/r, x)
```

Example 11.5 Find the Taylor expansion of the function $f(x) = \frac{1}{x-2}$ at $x = 0$ up to the 5-th order. Plot $f(x)$ and the 5-th order Taylor polynomial together.

Solution

```
# Find the Taylor expansion.
```

```
ftaylor:=taylor(1/(x-2), x = 0, 5)
```

```
# convert the Taylor series into a polynomial
```


```
fpoly:=convert(ftaylor, polynom)
```

```
# Plot the functions
```

```
plot([1/(x-2), fpoly], x=-1..1)
```

 **Exercise 11.4** Find the interval of convergence of the power series

$$\sum_{n=1}^{\infty} \frac{(-4)^n x^n}{\sqrt{n}}.$$

 **Exercise 11.5** Find the Taylor expansion of the function $f(x) = \sin x$ at $x = 0$ up to the 5-th order. Plot $f(x)$ and the 5-th order Taylor polynomial together over the interval $[-\pi, \pi]$.

11.3 Taylor Expansion

Let $f(x)$ be a function. Assume that the k -th order derivatives $f^{(k)}(a)$ exist for $k = 1, 2, \dots, n$. The polynomial

$$T_n(x) = \sum_{k=0}^n \frac{f^{(k)}(a)}{k!} (x-a)^k$$

is called the n -th degree Taylor polynomial of f at a .

Let $f(x)$ be a function has derivative at a up to all orders. Set

$$R_n(x) = \sum_{k=n+1}^{\infty} \frac{f^{(k)}(a)}{k!} (x-a)^k, \quad |x-a| < R,$$

which is called the reminder of the Taylor series

$$\sum_{k=0}^{\infty} \frac{f^{(k)}(a)}{k!} (x-a)^k.$$

If

$$\lim_{n \rightarrow \infty} R_n(x) = 0$$

for $|x-a| < R$, then $f(x)$ is the sum of the Taylor series on the interval, that is

$$f(x) = \sum_{k=0}^{\infty} \frac{f^{(k)}(a)}{k!} (x-a)^k, \quad |x-a| < R.$$

If $|f^{(n+1)}x| \leq M$ for $|x-a| \leq d$, then the reminder R_n satisfies the follow inequality

$$|R_n(x)| \leq \frac{M}{n+1} |x-a|^{n+1} \quad \text{for } |x-a| \leq d.$$

Roughly speaking, the absolute value of the reminder $|R_n(x)|$ determines how accurate the Taylor polynomial approximation.

Example 11.6 Approximate function $f(x) = \sin x$ by the degree 3 Taylor polynomial at $x = 1$.

Solution

```
# Find the Taylor series.
```

```
fTs:=taylor(sin(x), x = 0, 4)

# Convert the Taylor series into a polynomial

fTp:=convert(fTs, polynom)

# Evaluate the Taylor polynomial at 1

subs(x=1, fpolyapprox)
```

Example 11.7 Plot the function

$$g(x) = \begin{cases} e^{-\frac{1}{x^2}} & x \neq 0 \\ 0 & x = 0 \end{cases}$$

and its 5-th order Taylor polynomial over the domain $[-2..2]$. What can you conclude?

Solution

```
# Define a piece-wisely defined function.

g:=piecewise(x!=0, exp(-1/x^2), 0)

# Find Taylor polynomial of degree 5.


for n to 5 do T := (eval(diff(g(x), x$n), x = 0))*x^n/factorial(n)+T end do


# Plot the functions


plot([g,T],x=-2..2, color=[red, blue])
```

The graphs of the functions are shown in the picture.

In the solution, `x$n` is a shortcut option for `x, x, x, x, x` in the `diff` command.

 **Exercise 11.6** Approximate function $f(x) = e^x$ by the degree 5 Taylor polynomial at $x = 1$.

 **Exercise 11.7** Compare the function $y = \sin x$ with its degree 10 Taylor polynomial at $x = 0$.



A non-analytic smooth function